汇编语言的编程艺术(第2版)

第2章  数 据 表 示




数 据 表 示

在学习汇编语言的时候，许多初学者遇到的主要障碍都是二进制和十六进制数字系统的用法。虽然十六进制数与通常所用的数的确存在一些差异，但是它们的优点远远大于缺点。理解这些数字系统是很重要的，因为它们简化了许多复杂的问题，包括位运算、有符号数的表示、字符代码以及压缩数据。

本章将讨论几个重要的概念，其中包括：

二进制和十六进制数字系统

· 二进制数据的结构(位、半字节、字节、字以及双字)

· 二进制的有符号和无符号数字系统

· 二进制值的算术、逻辑、移位以及循环移位运算

· 位域和压缩数据

这些都是基础知识，本书剩下的部分将依赖于对这些概念的理解。如果您已经在其他课程或者研究中熟悉了这些术语，那么在学习第3章之前至少应该复习一下这些内容。如果您对它们还不熟悉，那么应该认真学习本章内容。本章的内容非常重要，千万不要跳过任何基础知识。

2.1  数字系统

现代的计算机系统大多数都不用十进制数字系统来表示数值。它们一般都采用二进制或者二进制补码数字系统。

2.1.1  回顾十进制系统

由于人们使用十进制数字系统的时间已经很长了，以至于都认为它是理所当然的。当看到数字“123”时，您不会去考虑数值123，而是在大脑中考虑该数值代表了多少项。实际上，数字123表示的是：


[image: image2.wmf]0

1

2

10

*

3

10

*

2

10

*

1

+

+


或者

100+20+3

在十进制位置数字系统当中，小数点左边的每一位数字都代表0~9之间的数值乘以10的递增次幂。小数点右边的数字都代表0~9之间的数值乘以10的递减负次幂。例如，数值123.456可表示为：

或者

100+20+3+0.4+0.05+0.006

2.1.2  二进制数字系统

现代的计算机系统大多都采用二进制逻辑进行操作。计算机使用两个电平(通常是0v和+2.4~5v)来表示数值。有了这两个电平，我们就正好可以表示两个不同的数值。它们可以是任意两个不同的数，但一般都表示0和1。这两个数同时对应于二进制数字系统中所使用的两个数字。

二进制数字系统与十进制系统的原理相同，但有两点差别：二进制只使用数字0和1(而不是0~9)，并且二进制使用2的幂而不是10的幂。因此，将二进制数转换为十进制数非常容易。二进制串中的每个“1”，都要加上2n，其中n是从0开始排序的二进制数字的位置数。例如，二进制数值110010102表示：

1*27 +1*26 +0*25 +0*24 +1*23 +0*22 +1*21 +0*20
=

128 +64 +8 +2

=

20210
将十进制数转换为二进制数要稍微复杂一些。必须要找到那些加到一起时得到十进制结果的2的幂数。

一种简单的方法叫做“偶/奇-除2”算法。该算法的步骤如下所示：

(1) 如果是偶数，就得到一个0；如果为奇数，就得到一个1。

(2) 用2来除这个数并舍弃小数部分或者余数。

(3) 如果商为0，算法就完成了。

(4) 如果商不是0而是奇数，就在现有的二进制串之前插入一个1；如果该数为偶数，那么就在二进制串之前附加一个0。

(5) 返回步骤(2)，再重复进行。

虽然在高级语言中，二进制数并不重要，但它们在汇编语言中却处处可见。所以您应该熟悉它们。

2.1.3  二进制格式

从最纯粹的意义上来说，每个二进制数都包含无数个数字(或位(bit)，它是二进制数字(binary digit)的简写)。例如，我们可以用下面任何一种方式来表示数字5：

101   00000101   0000000000101   …000000000000101

二进制数前面可以加任意多个前导0，而不会改变它的值。

我们遵守一个约定，如果在数值中出现了前导0，就将它们都忽略掉。例如，1012表示数值5，但是由于80x86一般都对8位数进行操作，我们发现用0将所有的二进制数扩展为4位或者8位的若干倍，处理起来会容易得多。因此根据这个约定，我们将数字5表示为01012或者000001012。

在美国，为了能使一些比较大的数更容易读，大多数人都将每三位用逗号隔开。例如，1,023,435,208就比1023435208读起来和理解起来更容易。在本书中我们将对二进制数采用相似的约定，将每四个二进制位分为一组并用下划线来分隔。例如，我们将二进制数1010111110110010写成1010_1111_1011_0010。

我们将采用下面的方式对每一位进行编号：

(1) 二进制数的最右一位为第0位。

(2) 每向左一位就给一个后继的位号。

8位二进制数使用位0~7：

X7X6X5X4X3X2X1X0
16位二进制数使用位0~15：

X15X14X13X12X11X10X9X8X7X6X5X4X3X2X1

32位二进制数使用位0~31，等等。

第0位是低位(L.O.)位(有人把这一位称为最低位)。最左边的位一般叫做高位(H.O.)位(或者最高位)。我们将用位号来引用中间的各个位。

2.2  十六进制数字系统

二进制系统中存在的一大问题就是冗长。数值20210需要8个二进制数字表示。如果采用十进制表示，3个数字就足够了。因此，采用十进制表示的数就要比采用二进制紧凑得多。这一事实并没有被设计二进制的工程师遗忘。当要处理一个较大的数值时，二进制数就变得非常不适用了。遗憾的是，计算机是使用二进制进行思考的，所以在大多数时间里，使用二进制数字系统会更加方便一些。虽然我们可以在十进制和二进制之间进行转换，但是这种转换并不简单。十六进制(基数为16)数字系统解决了二进制系统中固有的很多问题。十六进制数具有我们一直在寻求的两个特征：它们非常紧凑，并且将它们转换为二进制或者反过来转换都非常简单。因此，大多数计算机系统工程师都采用十六进制数字系统。

由于十六进制数的基数为16，所以十六进制小数点左边的每位数字表示该数值与16的逐次幂的乘积。例如，123416等于

或者

每个十六进制数字都可以代表0~1510这16个数字之间的某一个。由于只存在10个十进制数字，所以我们需要增加六个额外的数字来代表1010~1510这几个数。我们并没有为这些数创建新的符号，而是使用字母A~F来表示。下面全部都是有效的十六进制数：

由于我们经常需要向计算机系统输入十六进制数，所以需要采用不同的机制来表示十六进制数。在大多数计算机上，我们都不能使用下标来表示相关数值的基数。我们将采用如下的约定：

· 所有的十六进制数都以$字符开头，例如，$123A4。

· 所有的二进制数都以一个百分号(%)开头。

· 十进制数没有前缀。

· 如果从上下文可以很清楚地知道基数，本书可能会省掉$或者%符号。

下面是一些有效的十六进制数的示例：

$1234

$DEAD

$BEEF

$AFB

$FEED

$DEAF

正如您所看到的，十六进制数非常紧凑，且可读性较好。另外，还很容易进行十六进制与二进制之间的转换。考虑如下所示的表2-1，该表提供了十六进制数与二进制数相互转换时所需的全部信息。
表2-1  二进制/十六进制转换

	二  进  制
	十 六 进 制

	%0000
	$0

	%0001
	$1

	%0010
	$2

	%0011
	$3

	%0100
	$4

	%0101
	$5

	%0110
	$6

	%0111
	$7

	%1000
	$8



(续表)   

	二  进  制
	十 六 进 制

	%1001
	$9

	%1010
	$A

	%1011
	$B

	%1100
	$C

	%1101
	$D

	%1110
	$E

	%1111
	$F


要将某个十六进制数转换为二进制数，只需要将该数中的每一位十六进制数字替换为4位对应的二进制数。例如，要将$ABCD转换为二进制数，就只需要参照表2-1对每一位十六进制数字进行转换：

	A
	B
	C
	D
	十六进制

	1010
	1011
	1100
	1101
	二进制


将二进制数转换为十六进制格式也一样容易。第一步就是将二进制数补零，使得该数的位数变为4的倍数。例如，给定一个二进制数1011001010，第一步就是在该数的左边添加两位使得它包含12位数字。要转换的二进制数就是001011001010。下一步是将二进制数每四位分成一组，例如，0010_1100_1010。最后，在表2-1中查这些二进制数并用相应的十六进制数字来替换，也就是$2CA。将这一过程的难度与十进制与二进制之间的转换或者十进制与十六进制之间的转换进行对比！

由于经常需要在十六进制与二进制之间进行转换，所以最好花上一点时间牢记表2-1。即使有一个可以完成这种转换的计算器，您也会发现，在进行二进制与十六进制之间的转换时，手动转换会更快、更方便一些。

2.3  数据结构

单纯从数学上来说，一个数值可以取任意多位。而从另一方面来说，计算机一般都只能处理特定位数的值。一般有1位(bit)一组的、4位一组的(半字节，nibble)、8位一组的(字节，byte)、16位一组的(字，word)、32位一组的(双字，double word或dword)、64位一组的(四字，quad word或qword)、128位一组的(长字，long word或lword)等。长度并不是任意的。之所以用这些特定的值，是有充分理由的。本部分将讲述在Intel 80x86芯片中通常用到的位组。

2.3.1  位

在二进制计算机上，数据的最小单元是一位。用一位可以表示两个不同的数据项。例如，0或者1、真或者假、开或者关、男或者女、对或者错。但是，位并不局限于表示二进制数据类型(也就是说，那些只有两个不同数值的对象)。可以用一位来表示数字723与1 245，或者表示数值6 254与5。也可以用一位来表示红色与蓝色。甚至还可以用一位来表示两个不相关的对象。例如，可以用一位来表示红色与数字3 256。用一位可以表示任意两个不同的数值，但是一位只能表示两个不同的数值。

不同的位可以表示不同的对象。例如，可以用一个位表示数值0与1，而用另一个位表示真与假。那么该如何通过观察这些位来分辨它们呢？当然，答案是您不能分辨。但是这却阐明了计算机数据结构背后的所有思想：数据就是您所定义的对象。如果用一位来表示布尔值(true/false)，那么这一位就被定义为表示true或者false。要让一位具有真正的意义，就必须保持前后一致。如果在程序的某一点上，用一位来表示true或者false，那么在后面就不应该用它来表示红或者蓝。

由于试图要表示的内容大多数都需要多于两位，一位数值并不是使用得最多的数据类型。但是，由于其他数据类型都由多位构成的组所组成，所以位将在程序中扮演重要角色。当然，有几种数据类型需要两个不同的数值，所以看起来这些位本身也很重要。但是，您很快就会看到单独的位很难操作，所以我们通常都会用其他数据类型来表示具有两种状态的数值。

2.3.2  半字节

半字节是4位一组的数据类型。它并不是特别令人感兴趣，但其中的两种表示法很有用：BCD(binary-coded decimal，二进制编码的十进制数)数1和十六进制数。它用4位来表示一个BCD或者十六进制数字。用半个字节，我们可以表示小于16的不同数值，因为4位串有16种不同的组合：

	0000

	0001

	0010

	0011

	0100

	0101

	0110

	0111

	1000

	1001

	1010

	1011

	1100

	1101

	1110

	1111


对于十六进制数而言，数值0、1、2、3、4、5、6、7、8、9、A、B、C、D、E、F都用4位来表示。BCD只用到了10个不同的数字(0、1、2、3、4、5、6、7、8、9)，它也需要4位(因为用3位只能表示8个不同的数值，用4位可表示的另外6个值不在BCD表示法中使用)。事实上，任意16个不同的数值都可以用一个半字节来表示，但是十六进制和BCD数字是我们可以用半字节来表示的主要数据项。

2.3.3  字节

毫无疑问，80x86微处理器所使用的最重要的数据结构是字节。一个字节由8位组成。80x86的主存和I/O地址都是字节地址。这就意味着80x86程序能够独立访问的最小单元是8位数值。要访问更小的数需要读入包含该数据的字节，并屏蔽掉不需要的位。字节中的位通常都从0~7进行编号，如图2-1所示。

图2-1  位编号

位0是低位位或者称为最低位，位7是字节的高位位或者称为最高位。我们将根据编号来引用它们。

注意一个字节正好包含两个半字节(如图2-2所示)。

图2-2  一个字节中的两个半字节

位0~3构成了低位半字节，位4~7构成了高位半字节。因为一个字节正好包含两个半字节，所以字节数值需要两个十六进制数字。

因为一个字节包含8位，所以它可以表示28(即256)个不同的数值。一般来说，都使用一个字节来表示0~255内的数值、–128~+127内的有符号数(参阅2.8节)、ASCII/IBM字符代码以及其他一些不需要超过256个不同数值的专用数据类型。很多数据类型的数值都少于256个，所以8位通常已经足够了。

因为80x86是字节编址的机器，所以操作整个字节比操作位或者半字节的效率更高。因此，大多数程序员都用一个字节来表示那些不需要超过256个数值的数据类型，即使少于8位也足够了。例如，我们通常用000000012与000000002来分别表示布尔值true与false。

对于一个字节来说，最重要的用途在于保存字符数值。无论是键盘上敲入的字符、屏幕上显示的字符，还是打印机打印的字符，都具有数值。为了让它能与其他数值相通，PC一般都使用ASCII字符集的某种变体。ASCII字符集定义了128个字符代码。

因为字节是80x86存储空间中最小的存储单元，字节也正好是HLA程序中可以创建的最小变量；正如在第1章中所介绍的那样，可以用数据类型int8声明一个8位的有符号整型变量。因为int8对象是有符号的，所以可以用int8类型的变量表示–128~+127之间的数值。int8类型的变量中应该只存储有符号的数值；如果想要创建一个任意的字节变量，就应该使用byte数据类型，如下所示：

static



          byteVar:byte;

byte数据类型是一个部分非类型化数据类型。唯一与byte对象相关的类型信息就是它的长度(一个字节)。可以将任意一个8位数值(较小的有符号整数、较小的无符号整数、字符等)存储为字节变量。您需要自己跟踪存入字节变量的对象类型。

2.3.4  字

字是一个16位的组。我们将如图2-3中所示那样对一个字中的位从0~15进行编号。和字节的编号一样，位0是低位位。对于字来讲，位15是高位位。当要引用字中的其他位时，我们将使用它们的位置编号。

图2-3  字中的位编号

可注意到，一个字正好包含两个字节。位0~7组成了低位字节，位8~15组成了高位字节(见图2-4)。


[image: image15.wmf] 

高位字节

 

 

低位字节

 


图2-4  一个字中的两个字节

一个字也可以如图2-5那样进一步分解为4个半字节。0号半字节是字中的低位半字节，3号半字节是字中的高位半字节。我们将简单地称另两个半字节为“1号半字节”和“2号半字节”。

图2-5  字中的半字节

16位可以表示216 (即65536)个不同的数值。这些数值可以处于0~65535之间，或者像一般情况下那样，是–32768~+32767之间的有符号数值，或者是不超过65 536的任意其他类型的数值。字的三个主要用途分别是有符号短整数值、无符号短整数值以及Unicode字符。

字可以表示0~65535或者–32768~+32767之间的整数值。无符号数用与字中的位相应的二进制值来表示。有符号数使用二进制补码形式来表示(见2.8节)。作为Unicode字符，字可以表示多达65536个不同的字符，允许在计算机程序中出现非罗马字符集。和ASCII一样，Unicode是一种国际标准，允许计算机处理亚洲语、希腊语以及俄语字符这样的非罗马字符。

与字节相同，在HLA程序中也可以创建字变量。当然，在第1章中您已经看到了如何使用int16数据类型来创建16位的有符号整型变量。下面是一个使用word数据类型创建字变量的例子：

static



           w: word; 

2.3.5  双字

顾名思义，双字由两个字组成。因此，双字有32位长，如图2-6所示。

图2-6  双字中的位编号

双字可以被分成一个高位字和一个低位字、4个不同的字节或者8个不同的半字节(见图2-7)。

图2-7  双字中的半字节、字节和字

双字可以表示各种不同的内容。一般双字都用来表示32位整型值(允许表示0~4294967295之间的无符号数或者–2147483648~+2147483647之间的有符号数)。32位浮点数也可以用双字来表示。双字的另一个常见用法便是存储指针值。

在第1章中，已经看到了如何用int32数据类型来创建32位有符号整型变量。同样也可以用dword数据类型创建一个任意的双字变量，如下面的例子所示：

static



          d: dword;

2.3.6  四字和长字

显然，我们可以继续定义更长的类型。但是，80x86只支持特定的字长，所以没有理由再继续定义更长的类型。虽然字节、字以及双字是80x86程序中最普遍的字长，四字(64位)也是非常重要的，因为有些浮点数据类型需要64位。同样，现代80x86处理器的SSE/MMX指令集也可以对64位的数值进行操作。长字(128位)也很重要，因为后来的80x86处理器上的SSE指令集可以操作128位的数值。HLA允许用qword和lword类型声明64位和128位的数值，如下所示：

static

      q
     :qword;

      l
     :lword;

注意，也可以用如下所示的HLA声明定义64位和128位整型值：

static

i64

         :int64;

i128
         :int128;

但是请注意，不能用像mov、add以及sub这样的标准指令直接操作64位和128位的整型数，因为标准的80x86整数寄存器每次只处理32位。在第8章，将看到怎样操作这些扩展精度(extended-precision)的数值。

2.4  二进制数和十六进制数的算术运算


对二进制和十六进制数，我们可以执行若干种运算。例如，我们可以执行加法、减法、乘法、除法以及其他一些算术运算。虽然您不必是这方面的专家，但在一些紧急情况下应该能够用一张纸和一支笔手工完成这些运算。虽然说您应该能够手工完成这些运算，但正确的执行方法便是用计算器来做。市场上有几种这样的计算器，下面列出了十六进制计算器的一些制造商(2010年)：

· 卡西欧公司(Casio)

· 惠普公司(Hewlett-Packard)

· 夏普公司(Sharp)

· 德州仪器公司(Texas Instruments)

除此之外，其他的计算器制造商也能生产这样的计算器。其中，惠普生产的是最好的，但是他们的产品比其他厂家的更贵。夏普和卡西欧的产品售价都低于50美元。如果您打算什么样的汇编程序都做，那么拥有一款这样的计算器是很关键的。

要想明白您为什么应该花钱买计算器，请考虑下面的算术问题：

$9

+  $1

----

您可能认为这个问题的答案为$10。但这并不正确！正确答案是10，即$A，而不是16 ($10)。减法存在与此相似的问题：

   $10

-  $1

----

虽然真正的答案是$F，但您可能会认为答案为$9。记住，这个问题问的是“16与1的差”。当然，答案是15，即$F。

即使上面的两个问题没有干扰到您，但是当在考虑其他问题时，情急之下您的大脑还是会转回到十进制模式，就会产生不正确的结果。这个例子说明：如果必须手工用十六进制数进行算术运算，那您就需要放松心情认真做。或者可以先将数转换为十进制，再对十进制数进行运算，然后再将它们转换回十六进制。

2.5  关于数字及其表示

许多人都对数字及其表示产生了混淆。初学汇编语言的学生经常会提出这样一个问题：“EAX寄存器中有一个二进制数，怎样才能将它转换为EAX寄存器中的十六进制数呢？”答案是：“不用。”虽然存储器或者寄存器中的数以二进制表示存在很大争议，但是最好将存储在它们当中的数看做是抽象数量(abstract numeric quantity)。像128、$80或者%1000_0000这样的符号串并不是不同的数；它们只是同一个抽象数量的不同表示，即我们所说的“一百二十八”。在计算机中，数字是不考虑其表示方法的；唯一涉及表示的就是在以人类可读的格式输入或者输出该数值的时候。

数量的可读格式是字符串形式。要用人类可读的格式输出128，就必须将数值128转换为1后面跟2，2后面跟8的三字符序列。这就给出了数量的十进制表示。如果愿意，可以将数值128转换为三字符序列$80。数是相同的，但是我们将它转换成了不同的字符序列，因为我们可能想用十六进制而不是十进制方式表示该数。同样，如果我们想用二进制方式来表示这个数，那么就必须将该数值转换为一个1后面跟着七个0的串。

默认情况下，当调用stdout.put例程时，HLA以十六进制数字系统显示所有的字节、字、双字、四字以及长字变量。同样，HLA的stdout.put例程也以十六进制来显示所有的寄存器数值。考虑程序清单2-1中的程序，它将以十进制形式输入的值转换为它们的十六进制等效值。
程序清单2-1  十进制到十六进制的转换
program ConvertToHex;

#include( "stdlib.hhf" )

static

value: int32;

begin ConvertToHex;

stdout.put( "Input a decimal value:" );

stdin.get( value );

mov( value, eax );

stdout.put( "The value ", value, " converted to hex is $", eax, nl );

end ConvertToHex;
类似地，寄存器变量以及字节、字、双字、四字或者长字变量的默认输入也是十六进制。程序清单2-2中的程序与程序清单2-1中的功能相反：它输入十六进制数值，再以十进制形式输出。

程序清单2-2  十六进制到十进制的转换
program ConvertToDecimal;

#include( "stdlib.hhf" )

static

value: int32;

begin ConvertToDecimal;

stdout.put( "Input a hexadecimal value: " );

stdin.get( ebx );

mov( ebx, value );

stdout.put( "The value $", ebx, " converted to decimal is ", value, nl );

end ConvertToDecimal;
HLA的stdout.put例程选择了十进制作为int8、int16和int32变量的默认输出基数，但并不意味着这些变量都会保存十进制数。记住，存储器和寄存器保存的是数值，而不是十六进制数或者十进制数。stdout.put例程将这些数值转换为串，并输出得到的结果串。在HLA语言中，使用十六进制还是十进制进行输出只是一种设计选择而已。可以修改HLA，使得它将寄存器变量以及字节、字、双字、四字或者长字变量按十进制而不是十六进制输出。如果需要按十进制输出寄存器变量以及字节、字、双字、四字或者长字变量，那么简单地调用putiX例程中的一个就可以了。stdout.puti8例程按8位有符号整数类型输出其参数。任何8位参数都有效。所以可以将8位的寄存器变量、int8变量或者字节变量作为参数传递给stdout.puti8，其结果总是十进制形式的。stdout.puti16和stdout.puti32也为16位和32位对象提供了同样的功能。程序清单2-3展示了只使用寄存器EBX(也就是说，没有使用变量value)的十进制转换程序(与程序清单2-2是相对照的)。

程序清单2-3  没有使用变量的十六进制到十进制的转换
program ConvertToDecimal2;

#include( "stdlib.hhf" )

begin ConvertToDecimal2;

stdout.put( "Input a hexadecimal value: " );

stdin.get( ebx );

stdout.put( "The value $", ebx, " converted to decimal is " );

stdout.puti32( ebx );

stdout.newln();

end ConvertToDecimal2;
注意，HLA的stdin.get例程使用的默认基数与stdout.put所采用的默认基数相同。也就是说，如果要读入int8、int16或者int32类型的变量，默认的输入基数为十进制。如果要读入寄存器变量或者字节、字、双字、四字或长字变量，默认的输入基数则为十六进制。在读取寄存器变量或者字节、字、双字、四字或长字变量时，如果要将默认的输入基数改为十进制，那么可以调用stdin.geti8、stdin.geti16、stdin.geti32、stdin.geti64或stdin.geti128。

如果想按十六进制输入或者输出一个int8、int16、int32、int64或者int128类型的变量，那么可以调用stdout.puth8、stdout.puth16、stdout.puth32、stdout.puth64、stdout.puth128、stdin.geth8、stdin.geth16、stdin.geth32、stdin.geth64或stdin.geth128例程。stdout.put8、stdout.put16、stdout.put32、stdout.put64和stdout.put128例程分别按十六进制写8位、16位、32位、64位和128位的数值。stdin.geth8、stdin.geth16、stdin.geth32、stdin.geth64和stdin.geth128分别读8位、16位、32位、64位和128位的数值，并将它们返回在寄存器AL、AX或者EAX(或者在一个64位或者128位的参数位置)中。程序清单2-4展示了其中几个例程的使用。

程序清单2-4  stdin.geth32和stdout.puth32的示例
program HexIO;

#include( "stdlib.hhf" )

static

i32: int32;

begin HexIO;

stdout.put( "Enter a hexadecimal value: " );

stdin.geth32();

mov( eax, i32 );

stdout.put( "The value you entered was $" );

stdout.puth32( i32 );

stdout.newln();

end HexIO;
2.6  位逻辑运算

有4种基本的逻辑运算用于处理十六进制和二进制数：and、or、xor(异或)和not。与算术运算不同，要执行这些运算，并不一定需要十六进制的计算器。通常用手工计算比用电子设备更容易一些。逻辑and运算是二元(dyadic)2运算(意思是它正好有两个操作数)。这些操作数都是独立的二进制位。and运算就是：

0 and 0 = 0 

0 and 1 = 0

1 and 0 = 0

1 and 1 = 1

表示逻辑and运算的一种简洁方法是用真值表。真值表采用如表2-2所示的格式。

表2-2  and运算的真值表

	and
	0
	1

	0
	0
	0

	1
	0
	1


这就如同乘法表一样。左边一列值对应于and运算最左边的操作数。表中最上面一行的值对应于and运算最右边的操作数。对于特定的一对输入值，位于行列交汇处的值是那两个值一起进行逻辑and运算的结果。

用日常用语表述的话，逻辑and运算的意思是：“如果第一个操作数为1，并且第二个操作数也为1，结果就为1；否则结果为0。”我们也可以这样描述：“如果有一个或者两个操作数都为0，则结果为0。”

关于and运算，请注意一个重要的事实，可以使用它将结果强制为0。只要操作数中有一个为0，结果就为0而无须考虑其他操作数。例如，在上面的真值表中，标记为0的行中只包含0，标记为0的列中也只包含0。相反，如果一个操作数为1，那么结果就恰好是第二个操作数的值。and运算的这些结果是非常重要的，尤其是当我们想将某些位强制为0时。我们将在下一节中对逻辑and运算的用法深入研究。

逻辑or运算也是一个二元运算。它的定义是：
0 or 0 = 0

0 or 1 = 1

1 or 0 = 1
1 or 1 = 1
or运算的真值表采用了表2-3所示的格式。

表2-3  or运算的真值表

	or
	0
	1

	0
	0
	1

	1
	1
	1


通俗地讲，逻辑or运算就是：“如果第一个操作数或者第二个操作数(或者两个都)为1，那么结果为1；否则，结果为0。”这也被称为同或(inclusive-or)运算。

如果逻辑or运算的操作数中有一个为1，那么结果就总是为1，而不必考虑第二个操作数的值。如果一个操作数为0，结果就总是第二个操作数的值。与逻辑and运算相同，这是逻辑or运算一个很重要的附带结果，后面将证实它是很有用的。

注意，逻辑同或运算的意思与标准英语中的意思是有差异的。考虑这样一个句子：“我将要去商店或者我将要去公园”。这个句子暗示了说话者将要去商店或者去公园而不是两个地方都去。因此，英文or的意思与同或运算稍微有点不同；事实上，这是异或运算的定义。

逻辑xor(异或)运算也是一个二元运算。它的定义如下所示：

0 xor 0 = 0

0 xor 1 = 1

1 xor 0 = 1

1 xor 1 = 0
xor运算的真值表如表2-4所示。

表2-4  xor运算的真值表

	xor
	0
	1

	0
	0
	1

	1
	1
	0


通俗地说，逻辑xor运算的意思是：“如果第一个操作数或者第二个操作数为1，但不是两个都为1，那么结果就为1；否则结果就为0。”注意，异或运算的意思比逻辑or运算更接近于单词or。

如果逻辑异或运算的操作数中有一个值为1，那么结果就总是另一个操作数的相反值；也就是说，当有一个操作数为1时，如果另一个操作数也为1，结果就为0；如果另一个操作数为0，结果就为1。如果第一个操作数的值为0，那么结果就正好取第二个操作数的值。这个特征可以让您有选择地将位串中的某些位取反。

逻辑not运算是一元运算(意思是它只接受一个操作数)。如下所示：

not 0 = 1
not 1 = 0

表2-5为not运算的真值表。

表2-5  not运算的真值表

	not
	0
	1

	
	1
	0


2.7  二进制数和位串的逻辑运算

前一节定义的逻辑函数都是为单位操作数定义的。由于80x86使用的是由8位、16位或者32位构成的组，我们需要对这些函数的定义进行扩展以便对两位以及两位以上的数进行处理。80x86的逻辑函数是逐位进行操作的。给定两个值，这些函数对第0位进行操作就产生第0位的结果。它们对输入值的第1位进行运算就产生结果的第1位，依此类推。例如，如果要对下面两个8位数进行逻辑and运算，那么可以对每一列进行独立的逻辑and运算：

%1011_0101

%1110_1110

----------

%1010_0100
同样也可以将这种逐位进行的运算应用到其他逻辑函数中。

由于我们已经根据二进制数值定义了逻辑运算，您会发现对二进制数进行逻辑运算要比对其他表示方式进行运算容易得多。因此，如果要对两个十六进制数进行某种逻辑运算，那么应该首先将它们转换为二进制。这一点适用于大多数对二进制进行的基本逻辑运算(例如，and、or、xor等)。

在对位串(例如，二进制数)进行运算时，能够使用逻辑and/or运算将某些位强行置为0或者1，以及使用xor运算将某些位取反是非常重要的。这些运算允许您有选择地对位串中的某些位施加运算而不会影响其他位。例如，如果有一个8位二进制数X，想要保证第4~7位的值为0，那么可以将该值与二进制数%0000_1111进行逻辑and运算。这种按位进行的逻辑and运算会将X的高4位强行置为0，而保持低4位的值不变。同样，也可以通过将X与%0000_0001进行逻辑或运算，将X与%0000_0100进行逻辑异或运算来分别将X的最低位强行置为1，将X的第2位取反。以这种方式使用逻辑and、or以及xor运算来操作位串叫做屏蔽(masking)位串。之所以使用这个术语，是因为当我们要将某些位强行置为0、1或者对某些位取反时，可以用某个值(对于and用1，对于or/xor用0)屏蔽其他位。

80x86 CPU支持四条将这些按位进行的逻辑运算应用到它们的操作数上的指令。这些指令是and、or、xor以及not，其中and、or和xor指令使用的语法与add和sub指令的相同，如下所示：

and( source, dest );

or( source, dest );

xor( source, dest );
这些操作数与add操作数所受的限制相同。尤其是，source操作数必须是一个常量、存储器或者寄存器操作数，而dest操作数必须是存储器或寄存器操作数。而且，操作数的长度必须相等，还不能都是存储器操作数。这些指令下列通过等式进行按位的逻辑运算。

dest = dest operator source
80x86的逻辑not指令的语法稍有不同，因为它只有一个操作数。这个指令采用如下所示的形式：

not( dest );

该指令计算如下所示的结果：

dest = not( dest )

dest操作数必须是寄存器或者存储器操作数。该指令对指定的目标操作数中的所有位取反。

程序清单2-5中的程序接受用户输入的两个十六进制数，并计算它们的逻辑and、or、xor以及not值：

程序清单2-5  and、or、xor以及not的示例
program LogicalOp;

#include( "stdlib.hhf" )

begin LogicalOp;

stdout.put( "Input left operand: " );

stdin.get( eax );

stdout.put( "Input right operand: " );

stdin.get( ebx );

mov( eax, ecx );

and( ebx, ecx );

stdout.put( "$", eax, " and $", ebx, " = $", ecx, nl );

mov( eax, ecx );

or( ebx, ecx );

stdout.put( "$", eax, " or $", ebx, " = $", ecx, nl );

mov( eax, ecx );

xor( ebx, ecx );

stdout.put( "$", eax, " xor $", ebx, " = $", ecx, nl );

mov( eax, ecx );
not( ecx );

stdout.put( "not $", eax, " = $", ecx, nl );

mov( ebx, ecx );

not( ecx );

stdout.put( "not $", ebx, " = $", ecx, nl );

end LogicalOp;
2.8  有符号数和无符号数

到目前为止，我们一直都把二进制数当作无符号数。二进制数…00000表示0、…00001表示1、…00010表示2，依此类推直到无穷大。负数又怎样表示呢？有符号数在前面很多地方都用到了，我们也提到了二进制补码数字系统，但是还没有讨论怎样使用二进制数字系统表示负数。这正是本节所要讨论的内容。

为了用二进制数字系统表示有符号数，必须对数施加一个限制：它们的位数必须有限并且位数固定。为此，我们将把位数严格限制为8位、16位、32位、64位、128位或者其他一些较小的位数。

使用固定的位数只能表示一定数目的对象。例如，用8位只能表示256个不同的数值。与正数和0一样，负数本身就是对象；因此，我们必须从256个不同的8位数中拿出一部分来表示负数。换句话说，我们需要占用一些位组合来表示负数。为了公平起见，我们将从所有可能的组合中拿出一半来表示负数，另一半表示正数和0。所以可用一个8位字节来表示–128~–1内的负数和0~127内的非负数。用16 位的字可以表示–32 768~+32 767内的数值，用32位的双字可以表示–2 147 483 648~+2 147 483 647内的数值。一般情况下，用n位可以表示–2n-1~+2n-1–1内的有符号数。

好了，我们可以表示负数了。但到底应该怎样做呢？其实有很多方法，但是80x86微处理器采用补码形式来表示。在补码系统中，数字的最高位代表符号位(sign bit)。如果最高位为0，那么该数就是正的；如果最高位为1，那么该数就是负的。例如：

对于16 位数：

$8000是负数，因为最高位为1。
$100是正数，因为最高位为0。

$7FFF是正数。

$FFFF是负数。

$FFF($0FFF)是正数。

如果最高位为0，那么这个数就为正，并采用标准的二进制格式表示。如果最高位为1，那么这个数就为负数，并使用补码形式表示。要将一个正数转换为它的负数形式，即它的补码形式，可以使用下面的算法：

(1) 对该数的每一位取反—— 也就是说，应用逻辑not函数。

(2) 对取反后的结果加1，忽略最高位的溢出。

例如，计算5的8位等效值：

%0000_0101     5(二进制)
%1111_1010     对所有位取反

%1111_1011     加1以后得到结果

如果我们取–5，并对它进行补码操作，那么就可以得到原数值%0000_0101，这正符合我们的预期。

%1111_1011    –5的补码

%0000_0100    对所有位取反

%0000_0101    加1以后得到结果(+5)

下面的例子给出了一些正的和负的16位有符号数：
$7FFF: +32767，最大的16位正数

$8000: -32768，最小的16位负数

$4000: +16384
要将上面的数转换为它们相应的负数(也就是说，对它们取负值)，可以按照下面的方法进行：
$7FFF:






%0111_1111_1111_1111



+32767 

%1000_0000_0000_0000



对所有位取反 (8000h)

%1000_0000_0000_0001



加1(8001h 或 -32767)
$4000h:





%0100_0000_0000_0000



16384

%1011_1111_1111_1111



对所有位取反($BFFF)

%1100_0000_0000_0000



加1($C000 或 -16,384)
$8000:






%1000_0000_0000_0000



-32768

%0111_1111_1111_1111



对所有位取反($7FFF)

%1000_0000_0000_0000



加1(8000h 或-32768)

$8000 取反以后就变成了$7FFF。再加1就得到了$8000！这是怎么回事呢？–(–32768)与–32, 768相等吗？当然不相等了。但是由于16位有符号数不能表示32768，所以我们不能对最小的负数再取负。

为什么要对这么令人痛苦的数字系统进行纠缠呢？为何不把最高位用作符号标志位，再用剩下的几位来表示该数正的等效值呢(这叫做反码数字系统)？问题在于硬件。当出现这种情况时，对数值取负是最费事的。如果采用补码形式，大多数其他操作都变得与在二进制系统中一样容易。例如，要做加法5+(–5)，其结果为0。考虑一下当我们在补码系统中将这两个数相加时的情况：

%  0000_0101

%  1111_1011

------------
%1_0000_0000
结果是第9位为进位，而其余位都是0。当出现这种情况时，如果我们将最高位的进位忽略掉，那么当使用补码数字系统时，将两个有符号数相加就能够得到正确的结果了。这意味着对于有符号数和无符号数的加法和减法，我们可以使用相同的硬件。但是，如果使用其他的数字系统就不行。

通常，补码操作不需要手工完成。80x86微处理器提供了一条neg(negate)指令来完成这一操作。而且，只要按符号键(+/–或者CHS)，所有的十六进制计算器就都能执行该操作。不过，用手工进行补码操作也很容易，您应该知道怎样做。

另外，应该注意用一组二进制位表示的数据完全依赖于上下文。8位二进制数值%1100_0000可以表示一个字符、可以表示无符号十进制数值192，或者表示有符号数–64。作为一名程序员，您应该负责定义数据的格式，然后一致使用。

80x86的取负指令neg所使用的语法与not指令相同；也就是说，它只有一个目的操作数：

neg( dest );

该指令完成dest =–dest;运算，并且其操作数所受的限制与not所受的限制相同(它必须是一个存储单元或者寄存器)。neg可以对字节、字、双字对象进行操作。当然，由于这是一个有符号的整数操作，所以它只能对有符号的整数进行操作。程序清单2-6中的程序展示了用neg指令完成的补码操作：

程序清单2-6  补码操作的示例
program twosComplement;

#include( "stdlib.hhf" )

static

PosValue: int8;

NegValue: int8;

begin twosComplement;
stdout.put( "Enter an integer between 0 and 127: " );

stdin.get( PosValue );

stdout.put( nl, "Value in hexadecimal: $" );

stdout.puth8( PosValue );

mov( PosValue, al );

not( al );

stdout.put( nl, "Invert all the bits: $", al, nl );

add( 1, al );

stdout.put( "Add one: $", al, nl );

mov( al, NegValue );

stdout.put( "Result in decimal: ", NegValue, nl );

stdout.put

(

nl,

"Now do the same thing with the NEG instruction: ",

nl

);

mov( PosValue, al );

neg( al );

mov( al, NegValue );

stdout.put( "Hex result = $", al, nl );

stdout.put( "Decimal result = ", NegValue, nl );

end twosComplement;
正如前面所看到的一样，可以用int8、int16、int32、int64和int128数据类型来保存有符号的整型变量。您也看到了像stdout.puti8和stdin.geti32这样用于读写有符号整数值的例程。因为本节已经讲得非常清楚，必须对程序中有符号数和无符号数的计算进行区分，可能此时您正在问自己：“我应该如何声明和使用无符号整型变量呢？”

该问题的第一部分“如何声明无符号变量”最容易回答。当声明变量的时候，只需要简单地使用数据类型uns8、uns16、uns32、uns64 和uns128就可以了，例如：

static

u8: 
  uns8;
u16:  
uns16;

u32:  
uns32;

u64:  
uns64;

u128: 
uns128;
至于这些无符号变量的使用，HLA标准库为无符号变量的读取和显示提供了一组输入/输出例程。正如您猜到的一样，这些例程包括stdout.putu8、stdout.putu16、stdout.putu32、stdout.putu64、stdout.putu128、stdout.putu8Size、stdout.putu16Size、stdout.putu32Size、stdout.putu64Size、stdout.putu128Size、stdin.getu8、stdin.getu16、stdin.getu32、stdin.getu64和stdin.getu128。除了这些例程用于处理无符号数值以外，它们的使用方法与处理有符号整型数的相应例程一样。程序清单2-7中的源代码不但展示了无符号数的I/O操作，还展示了在同一个计算中混用有符号和无符号操作时出现的情况。

程序清单2-7  无符号数的I/O操作
program UnsExample;

#include( "stdlib.hhf" )

static

UnsValue: uns16;

begin UnsExample;

stdout.put( "Enter an integer between 32,768 and 65,535: " );

stdin.getu16();

mov( ax, UnsValue );

stdout.put

(

"You entered ",

UnsValue,

". If you treat this as a signed integer, it is "

);

stdout.puti16( UnsValue );

stdout.newln();

end UnsExample;
2.9  符号扩展、零扩展、压缩和饱和

由于采用补码形式表示的整数具有固定的长度，这样就出现了一个小问题。如果要将一个8位二进制补码数值转换成16位数会怎样呢？这个问题以及它的逆问题(将16位数转换成8位数)可以通过符号扩展(sign extension)和压缩(contraction)操作来实现。

考虑一下数值–64。该数的8位二进制补码为$C0，它的16位等效值为$FFC0。现在来考虑数值+64。该数的8位和16位等效值分别为$40和$0040。8位和16位数之间的差别可以用这样一条规则来描述：“如果该数为负，那么它的16位数字表示中高位字节就应该为$FF；如果该数为正，那么高位字节就应该为0。” 

将一个有符号数从几位扩展成更多位的数字表示是很容易的，只须将它的符号位复制到新格式的所有附加位中就可以了。例如，要将一个8位数符号扩展成为16位数，只要简单地将该数的第7位数字复制到16位数的第8~15位中就可以了。如果要将一个16位的数符号扩展成为双字，那么将第15位复制到双字的第16~31位中就可以了。

当要对长度可变的有符号数进行操作时，必须采用符号扩展的方法。您经常需要将一个字节与一个字相加。那么在进行该操作以前必须先将字节符号扩展成为一个字。其他的操作(尤其是乘法和除法)可能需要符号扩展成为32位：

符号扩展:

8位





16位





32位

$80





$FF80




$FFFF_FF80

$28





$0028




$0000_0028

$9A





$FF9A




$FFFF_FF9A

$7F





$007F




$0000_007F









$1020




$0000_1020

 







$8086




$FFFF_8086
要将一个无符号数扩展成为更长的数时，必须对该数进行零扩展。零扩展非常容易：只要将0保存到更长操作数的高位字节中就可以了。例如，要将一个8位数值$82零扩展成为16位，只需要简单地在高位字节上加0就可得到$0082。

零扩展:

8位





16位





32位

$80





$0080




$0000_0080

$28





$0028




$0000_0028

$9A





$009A




$0000_009A

$7F





$007F




$0000_007F








$1020    

$0000_1020

     




$8086    

$0000_8086
80x86提供了一些指令可将较短的数符号扩展或者零扩展成为较长的数。表2-6列出了一组用于对寄存器AL、AX和EAX进行符号扩展的指令。

表2-6  用于扩展AL、AX和EAX的指令

	指    令
	解    释

	cbw();
	通过符号扩展将AL中的字节转换为AX中的字

	cwd();
	通过符号扩展将AX中的字转换为DX:AX中的双字

	cdq();
	通过符号扩展将EAX中的双字转换为EDX:EAX中的四字

	cwde();
	通过符号扩展将AX中的字转换为EAX中的双字


注意，指令cwd(将字转换为双字)并不能将AX中的字符号扩展为EAX中的双字。而是将符号扩展后的高位字存储到DX寄存器中(符号DX:AX告诉您有一个双字，它的高16位保存在DX中，低16位保存在AX中)。如果想让AX符号扩展为EAX，就应该使用cwde指令(将字扩展成为双字)。

上面四条指令是您第一次看到的没有操作数的指令。这些指令的操作数由指令自身隐含。

在下面几章中，您将看到这些指令的重要性，并了解cwd和cdq指令要使用DX和EDX寄存器的原因。但是对于简单的有符号扩展操作来说，这些指令存在几个主要的缺点：不能指定源操作数和目的操作数，而且这些操作数都必须是寄存器。

对于一般的符号扩展操作来说，80x86提供了mov指令的扩展指令movsx(带符号扩展的移动)，在对数据进行复制的同时可对它进行符号扩展。movsx指令的语法与mov指令非常相似：

movsx( source,dest );

该指令与mov指令最大的区别在于目的操作数的数据宽度必须大于源操作数的数据宽度。也就是说，如果源操作数是一个字节，那么目的操作数就必须是字或者双字。同样，如果源操作数是字，目的操作数就必须是双字。另一个区别是目的操作数必须是寄存器；而源操作数可以是存储单元3。movsx指令不接受常量操作数。

对数值进行零扩展可以使用movzx指令。它的语法和所受的限制与movsx指令的相同。将8位寄存器(AL、BL、CL和DL)零扩展为与之对应的16位寄存器，可以通过向高位寄存器(AH、BH、CH或者DH)中载入0来实现。显然，如果要将AX零扩展为DX:AX或者将EAX零扩展为EDX:EAX，要做的就是在DX或者EDX中载入04。

程序清单2-8中的示例程序展示了符号扩展指令的应用。

程序清单2-8  符号扩展指令
program signExtension;

#include( "stdlib.hhf" )

static

i8:


int8;

i16:

int16;

i32:

int32;
begin signExtension;

stdout.put( "Enter a small negative number: " );

stdin.get( i8 );

stdout.put( nl, "Sign extension using CBW and CWDE:", nl, nl );

mov( i8, al );

stdout.put( "You entered ", i8, " ($", al, ")", nl );

cbw();

mov( ax, i16 );

stdout.put( "16-bit sign extension: ", i16, " ($", ax, ")", nl );

cwde();

mov( eax, i32 );

stdout.put( "32-bit sign extension: ", i32, " ($", eax, ")", nl );

stdout.put( nl, "Sign extension using MOVSX:", nl, nl );

movsx( i8, ax );

mov( ax, i16 );

stdout.put( "16-bit sign extension: ", i16, " ($", ax, ")", nl );

movsx( i8, eax );

mov( eax, i32 );

stdout.put( "32-bit sign extension: ", i32, " ($", eax, ")", nl );

end signExtension;
符号压缩是将一个数转换成一个位数更少、但值相同的数，该操作要稍微麻烦一些。符号扩展是不会失败的。给定一个m位的有符号数，使用符号扩展总是可以把它转换成n位的数(这里的n > m)。遗憾的是，当给定一个n位的数时，如果m < n，它就并不总能转换为m位的数。例如，考虑一下数值–448。作为一个16位的有符号数，它的十六进制表示为$FE40。但对于8位数值表示法而言，该数值太大了，所以它不能用8位来表示。这是转换过程中发生上溢的例子。

为了能正确地将一个值符号压缩为另一个值，必须看一下要删除的高位字节。要删除的高位字节只能全都是0或者$FF。如果遇到其他值，在不发生上溢的情况下就无法对它进行符号压缩。最后，结果数值的高位位必须与从该数中删除的每一位相匹配。这里有一些从16位压缩为8位的例子：

$FF80可以被符号压缩为$80。

$0040可以被符号压缩为$40。

$FE40不能被符号压缩为8位。

$0100不能被符号压缩为8位。

减小整数长度的另一种方法是用饱和(saturation)来实现。当必须将某个较长的对象转换成较短的对象，并且愿意接受可能的精度损失时，就可以使用饱和操作。用饱和操作对数值进行转换时，如果较大对象没有超出较小对象的表示范围，那么只需要简单地将较大的数复制到较小的数中即可。如果较长数值超出了较短数值的范围，就将它设置为较短对象范围内最大(或者最小)的数值来对该数值进行修剪(clip)。

例如，将一个16位的有符号整型数转换为8位的整型数时，如果该16位数值在–128~+127内，只须简单地将该16位数的低位字节复制到8位数中。如果该16位数的值大于+127，那么就要将该值修剪为+127，并将+127存入8位数中。同样，如果它的值小于–128，那么就要将最终的8位数修剪为–128。将32位的数修剪成为较短的数时，饱和操作的方法相同。如果较长的数值超出较短的数所能表示的范围，就将较短的数简单地设置为该数所能表示的最接近边界的数值。

显然，如果数据宽度较大的数值超出了数据宽度较小数值的范围，那么在进行转换时就会有精度的损失。虽然我们都不希望将数值修剪为较小的界值，但比起引发一次异常，或者拒绝进行计算，有时候这还是可以接受的。对于很多应用，比如音频或者视频处理，修剪后的结果还是可以识别的，所以这是一种合理的转换。

2.10  移位和循环移位

应用于位串的另一组逻辑操作是移位(shift)操作和循环移位(rotate)操作。这两类操作可以进一步细分为左移(left shift)、循环左移(left rotate)、右移(right shift)以及循环右移(right rotate)。这些操作对汇编语言程序员来说是非常有用的。

左移操作将位串中的每一位向左移动一个位置(图2-8给出了一个8位移位操作的例子)。

图2-8  左移操作

第0位移到第1位所在的位置，以前位于第1位的值被移动到第2位，依此类推。当然，自然会出现两个问题：“什么数会进入第0位？”和“最高位将移到哪里去？”我们将向第0位移入一个0，而最高位中以前的值将变成进位值。

80x86提供了一个左移指令shl来完成该操作。shl指令的语法为：

shl( count, dest );

操作数count要么是CL，要么是0~n内的一个常数，其中n比目的操作数的位数小1(也就是说，对于8位的操作数n=7；对于16位的操作数n=15；对于32位的操作数n=31)。操作数dest是一个典型的目的操作数；它可以是存储单元或者寄存器。

当操作数count为常数1时，shl指令完成如图2-9所示的操作。

图2-9  左移操作

在图2-9中，C表示进位标志位。也就是说，从操作数中移出的最高位被移入进位标志位。因此，指令shl( 1, dest );执行完以后可以通过立即对进位标志位进行检测来判断是否发生上溢(例如，可以使用if( @c ) then . . .或者if( @nc ) then . . .)。

Intel的说明文档指出，如果移动的位数不为1，那么进位标志位的状态就没有定义。进位标志位通常为移出目的操作数的最后一位，但是Intel并不保证这一点。

请注意，将一个数向左移就等同于将它与基数相乘。例如，将一个十进制数向左移一位(在数的右端加一个0)等效于将它乘以10(基数)：

1234 shl 1 = 12340   

其中，指令shl 1的意思是向左移动一位。因为二进制数的基数为2，将它向左移一位就相当于将它乘以2。如果要将一个二进制数左移两位，就用2与它相乘两次(也就是说，让它与4相乘)。如果要将一个二进制数向左移3位，就用8与之相乘(2*2*2)。一般来说，如果要将一个数左移n位，就将它乘以2n。

除了数据移动的方向以外，右移操作与左移操作相同。对于一个字节数，第7位移入第6位，第6位移入第5位，第5位移入第4位，依此类推。在进行右移的时候，我们将0移入第7位，第0位将作为进位位被移出(见图2-10所示)。

图2-10  右移操作

正如您所期望的，80x86提供了shr指令来完成目的操作数的右移操作。它的语法与shl指令相同，当然，只是您指定的是shr而不是shl。
shr( count,dest );

该指令向目的操作数的最高位移入0；将其他的每一位向右移动一个位置(也就是说，从较高位移向较低位)。最后，第0位被移入进位标志位中。如果将count定义为1，那么shr指令就进行如图2-11所示的操作。

图2-11  右移操作

Intel的说明文档中也指出，当移动的位数超过1位时，进位位就为未定义状态。

因为左移等效于乘2，那么右移就应该粗略地等效于除以2(或者一般情况下，除以该数的基数)。如果右移n位，那么就相当于除以2n。

关于除法，右移操作存在一个问题：右移只能等效于除数为2的无符号除法。例如，如果将254($FE)的无符号表示形式向右移动一位，就得到127($7F)，这符合预期。但是，如果将–2($FE)的二进制表示向右移动一位，就会得到127($7F)，这是错误的结果。之所以出现这个问题，是因为我们向第7位移入了一个0。如果第7位以前的值为1，那么这样做就会把负数变成正数。当除以2时，结果就不对了。
为了能够使用右移操作来完成除法运算，我们必须再定义一个移位操作：算术右移。5算术右移操作与普通的右移操作(逻辑右移)相同，只有一点例外：算术右移操作并不向最高位移入0，而是将最高位复制回自身；也就是说，在移位的时候并不对最高位进行修改，如图2-12所示。

图2-12  算术右移操作

算术右移操作一般都能够产生希望的结果。例如，如果对–2($FE)进行算术右移操作，就可以得到–1($FF)。但是关于算术右移操作请记住一点，该操作总是将数值四舍五入为小于或者等于实际结果的最接近的整数值。基于高级语言编程的经验和截取整数的标准规则，大多数人都认为这意味着除法的结果将总是被舍入为0。但事实并非如此。例如，如果对–1($FF)进行算术右移，结果是–1而不是0。因为–1小于0，所以算术右移操作将它四舍五入为–1。这并不是算术右移操作中出现了错误；它只是使用了与整数除法不同(虽然有效)的定义。
80x86提供了算术右移指令sar(shift arithmetic right)。该指令的语法大致与shl和shr相同。其语法为：

sar( count,dest );

对于操作数count和dest的限制通常对它都适用。当count为1时，该指令的操作如图2-13所示。

图2-13  sar(1, dest)操作

Intel的说明文档中提出当移动的位数超过1位时，进位位处于未定义状态。

另一对有用的操作是循环左移和循环右移。这两种操作与右移和左移操作基本相同，但存在一个主要的差别：从一端移出的位会在另一端被移回。图2-14对这两种操作进行了演示。

图2-14  循环左移和循环右移操作

80x86提供了rol(循环左移)和ror(循环右移)指令来对它们的操作数完成这两种基本的操作。这两条指令的语法与移位指令相似：

rol( count,dest );
ror( count,dest );

同样，当移动的位数为1时，这些指令也提供了特殊的操作。在此条件下，这两条指令也将移出目的操作数的位复制到进位标志位中，如图2-15和图2-16所示。

图2-15  rol(1, dest)操作

请注意，Intel的说明文档提出当循环移动的位数超过1位时，进位标志位将处于未定义状态。

图2-16  ror(1, dest)操作

对于循环移位操作来说，让移位操作中的输出位通过进位标志位移动，并将先前进位标志位的值移回移位操作的输入位通常会更方便一些。80x86的进位左循环移位指令rcl和进位右循环移位指令rcr可以达到这一目的。这两种指令的语法如下所示：

rcl( count,dest );
rcr( count,dest );

与其他的移位操作和循环移位操作相同，操作数count是一个常数或者CL寄存器，并且目的操作数dest为一个存储单元或者寄存器。操作数count的数值必须小于目的操作数dest的位数。对于count数值为1的情况，这两条指令完成如图2-17所示的循环移位。

图2-17  rcl(1, dest)和rcr(1, dest)操作

Intel的说明文档提出当循环移动的位数超过1位时，进位标志位处于未定义状态。

2.11  位域和压缩数据

虽然80x86在对字节、字以及双字数据类型进行处理时效率最高，但偶尔也需要对8位、16位、32位之外的数据类型进行处理。例如，考虑格式为04/02/01的日期。其中用了三个数值来表示这个日期：月份、天数以及年份。当然，月份使用的数值为1~12。至少需要4位(最多16个数值)来表示月份。天数的范围为1~31，所以需要5位(最多32个不同的数值)来表示。假设我们用0~99来表示年份，它需要7位(最多可以表示128个不同的数值)来表示。4加5加7等于16位，即两个字节。换句话说就是我们可以将日期数据压缩到两个字节，而不像我们分别用一个字节来表示月份、天数和年份，那样需要三个字节。这样在存储每个日期的时候就节省了一个字节，如果有大量的日期需要存储，就可以节省更多的空间。这些位可以像图2-18所示那样分布。

图2-18  短压缩数据格式(两字节)

MMMM表示组成月份的4位数；DDDDD表示组成天数的5位数；YYYYYYY表示组成年份的7位数。每组表示一个数据项的位的集合就是一个位域(bit field)。例如，2001年4月2日表示为$4101:

 01

虽然压缩的数值是有空间效率的(也就是说，就存储器的使用来说非常有效)，但在进行计算时效率却非常低(比较慢)。那么是什么原因呢？这是因为它需要额外的指令解压缩不同位域中的数据。这些额外的指令需要消耗额外的时间来执行(并且还需要额外的字节来存储指令)；因此，必须仔细地考虑压缩后的数据域是否能够达到节省的目的。程序清单2-9中的示例程序演示了在对这个16位的日期格式进行压缩和解压缩的时候需要完成的工作。

程序清单2-9  压缩和解压缩日期数据
program dateDemo;

#include( "stdlib.hhf" )

static
day:



uns8;

month:

uns8;

year:


uns8;

packedDate: word;

begin dateDemo;

stdout.put( "Enter the current month, day, and year: " );

stdin.get( month, day, year );

// Pack the data into the following bits:

//

//
15

14

13

12

11

10

9

8

7

6

5

4

3

2

1

0

//

m


m


m


m


d


d

d

d

d

y

y

y

y

y

y

y

mov( 0, ax );

mov( ax, packedDate ); // Just in case there is an error.

if( month > 12 ) then

stdout.put( "Month value is too large", nl );

elseif( month = 0 ) then

stdout.put( "Month value must be in the range 1..12", nl );

elseif( day > 31 ) then

stdout.put( "Day value is too large", nl );

elseif( day = 0 ) then

stdout.put( "Day value must be in the range 1..31", nl );

elseif( year > 99 ) then

stdout.put( "Year value must be in the range 0..99", nl );

else

mov( month, al );

shl( 5, ax );

or( day, al );

shl( 7, ax );

or( year, al );

mov( ax, packedDate );

endif;

// Okay, display the packed value:

stdout.put( "Packed data = $", packedDate, nl );
// Unpack the date:

mov( packedDate, ax );

and( $7f, al );









// Retrieve the year value.

mov( al, year );

mov( packedDate, ax );


// Retrieve the day value.

shr( 7, ax );

and( %1_1111, al );

mov( al, day );

mov( packedDate, ax );

// Retrieve the month value.

rol( 4, ax );

and( %1111, al );

mov( al, month );

stdout.put( "The date is ", month, "/", day, "/", year, nl

end dateDemo;
当然，在经历“千年虫”的问题后，现在如果再采用一种限制在100年(或者甚至于127年)以内的日期格式是非常愚蠢的。如果考虑到软件从现在算起能运行100年，那么采用三字节的格式可能比采用两字节的格式要明智一些。但是在关于数组的章节中将会看到，应该尽可能地创建长度为2的偶数次方的数据对象(单字节、两字节、四字节、八字节等)，否则就会带来性能损失。因此，也许采用四字节将该数据压缩成一个双字变量是一种比较明智的做法。图2-19给出了一个四字节日期的数据组织方式。

图2-19  长压缩数据格式(四字节)
在这种长压缩数据格式中，除了对表示年份的位数进行扩展以外，还做了几点变动。首先，因为在32位双字变量中存在额外的位，这种格式将额外的位分配给了月份域和天数域。因为这两个域现在分别都由8位组成，所以它们可以作为一个字节对象很容易从双字中提取。这样给年份域留下的位数就少了，但是表示65536年应该足够了；您几乎可以确定从现在算起63000年后，即这一日期格式不再有效的时候，您的软件不可能还在使用。

当然，您可能会争辩说这并不是一种压缩的日期格式。毕竟，我们需要三个数值，其中有两个正好能放到一个字节中，而一个可能至少需要两个字节。因为这种“压缩”的数据格式同没有压缩的版本一样都需要占据四个字节，那么这种格式有什么特别之处呢？您会注意到长压缩数据格式与图2-18中的短压缩数据格式之间的另一个差别就是，事实上这种长压缩数据格式对所有的位进行了重新分配，所以年份域到了高位位，月份域处于中间位，而天数域处于低位位。这种排列是非常重要的，因为它使得两个日期很容易进行比较，以判断某个日期是小于、等于还是大于另一个日期。考虑下面的代码：

mov( Date1, eax );







// Assume Date1 and Date2 are dword variables

if( eax > Date2 ) then




// using the Long Packed Date format.
<< Do something if Date1 > Date2 >>
endif;
如果将日期中的各个域分别保存到不同的变量中或者对各个域进行不同的结构划分，就不能用这么直接的方式来对Date1和Date2进行比较了。因此，即使没有节省空间，这个例子仍然阐明了对数据进行压缩的另一个原因：它可以使某些运算更加方便或者更加有效(正好与通常进行数据压缩的情况相反)。

在实际应用中，压缩数据类型的例子有很多。可以将8个布尔数值压缩到一个字节中；可以将两个BCD数压缩到一个字节中，等等。当然，EFLAGS寄存器是压缩数据的一个典型例子(见图2-20)。该寄存器将9个重要的布尔对象(加上7个重要的系统标志)压缩到一个16位的寄存器中。其中一些标志需要经常访问。正由于这个原因，80x86指令集提供了很多方法对EFLAGS寄存器中的各个位进行操作。当然还可以在if语句或者其他使用布尔表达式的语句中使用HLA的@c、@nc、@z、@nz等伪布尔变量来测试很多条件码标志。

除了条件码

以外，80x86还提供了对某些标志直接进行处理的指令(见表2-7)。

表2-7  影响某些标志的指令

	指    令
	解    释

	cld()；
	清除方向标志(设置为0)

	std()；
	设置方向标志(设置为1)

	cli()；
	清除中断禁止标志

	sti()；
	设置中断禁止标志

	clc()；
	清除进位标志

	stc()；
	设置进位标志

	cmc()；
	将进位标志取反

	sahf()；
	将AH寄存器存储到EFLAGS寄存器的低8位中

	lahf()；
	从EFLAGS寄存器的低8位加载AH


还有其他一些能对EFLAGS寄存器产生影响的指令；但是这些指令阐明了如何对EFLAGS寄存器中的几种压缩布尔数值进行访问。特别是lahf和sahf指令为把EFLAGS寄存器的低8位当作一个8位字节来访问提供了一种很便利的途径。图2-20为EFLAGS寄存器的布局图。

图2-20  EFLAGS寄存器的布局

lahf(用EFLAGS寄存器的低8位装载AH寄存器)和sahf(将AH寄存器的EFLAGS存储到布局寄存器的低8位中)这两条指令的语法如下所示：

lahf();

sahf();

2.12  浮点运算简介

整数运算不允许出现小数值。因此，现在的CPU都支持一种实数运算的近似运算：浮点运算。浮点运算存在的一个大问题就是它不服从代数学的标准规则。但是，在进行浮点运算时，很多程序员都采用标准的代数学规则。这就是很多程序发生错误的原因。本节的一个主要目标就是对浮点运算所受到的限制进行描述，使您明白如何正确使用它。

标准的代数学规则只适用于无限精度(infinite precision)的运算。考虑一个简单的语句x：=x+1，其中x是一个整数。只要不发生上溢，在现在所有的计算机上，该语句都遵循代数学的标准规则。也就是说，该语句只对x的某些值有效(最小值<= x <最大值)。大多数的程序员对此并没有疑问，因为他们都知道这样一个事实，即程序中的整数并不遵循标准的代数学规则(例如，5/2
[image: image32.wmf]¹

)。

整数不服从标准规则是因为计算机只能用有限的位数来表示它们，所以任何大于最大值或者小于最小值的(整数)数值都表示不出来。浮点数也存在同样的问题，而且情况只会更糟。毕竟，整数是实数的子集。因此，浮点数必须同样能表示整数的无限集合。但是，在任意两个整数之间都存在无限个实数数值，所以这个问题就糟糕透顶了。因此，不但必须将数值限制在最大值和最小值的范围内，而且并不是处于这个范围内的所有数值都能够表示。

为了表示实数，大多数的浮点格式都采用科学记数法，用其中的某几位来表示尾数，再用更少的位来表示指数。最后的结果是浮点数只能表示有效数字的个数有限的数值。这对浮点运算有很大的影响。为了看到有限精度运算所带来的影响，我们在例子中将采用一种简化了的十进制浮点格式。我们的浮点格式由带三个有效数字的尾数和一个两位的十进制指数组成。尾数和指数都是有符号数，如图2-21所示。

图2-21  浮点格式

在科学记数法中，对两个数进行加法或者减法操作时必须对这两个数进行调整，使得它们的指数相同。例如，当1.23e1与4.56e0做加法时，必须对它们的数值进行调整，使它们具有相同的指数。一种方法就是将4.56e0转换成0.456e1，然后做加法。这样就得到结果为1.686e1。遗憾的是，该结果不能满足必须包含三个有效数字这一限制，所以必须将结果四舍五入(round)或者截取(truncate)得到三个有效数字。四舍五入一般能产生最正确的结果，所以对结果进行四舍五入以后就得到1.69e1。正如您所看到的，精度(我们在计算中保留的数字或者位的数目)不足会对精确性(计算的正确性)产生影响。

在前面的例子中，我们之所以对结果进行四舍五入，是因为在进行计算的时候，我们保留了四位有效数字。如果在进行运算时我们的浮点运算受到限制，只能保留三位有效数字，那么就不得不将小数部分的最后一位数字截取掉得到1.68e1，它就更加不精确了。为了提高浮点运算的精度，有必要在进行计算时增加额外的数字。在运算过程中所增加的额外数字被称为保护数字(guard digit)(或者在二进制格式中称为保护位)。在运算环节中，它们极大地提高了精度。

运算中的精度损失通常不值得担心，除非非常关心运算的精度。但是如果对一组浮点操作进行计算，错误就会发生累积，并对运算本身产生很大的影响。例如，假设要对1.23e3与1.00e0做加法。首先对这两个数进行调整使得它们的指数相同，然后做加法1.23e3 + 0.001e3。即使在进行四舍五入之后，这两个数的和也是1.23e3。对于您来说，看起来这似乎是完全合理的；毕竟只能保留三位有效数字，因此对一些较小的数值进行加法时根本就不会影响结果。但是，我们假设要对1.23e3累加十次1.00e0。第一次将1.00e0和1.23e3相加得到1.23e3。同样，第二次、第三次、第四次、……、第十次将1.00e0和1.23e3相加时，我们都会得到相同的结果1.23e3。另一方面，如果将1.00e0自身累加十次，然后再将结果(1.00e1)和1.23e3相加，那么应该会得到一个不同的结果1.24e3。对有限精度运算有一定的了解是非常重要的：

求值的顺序会对结果的精度造成影响。
在对浮点数进行加法或者减法操作的时候，如果其中一个数的数量级(即指数)接近于另一个数，那么就能得到更加精确的结果。如果要执行一个涉及加法和减法操作的运算链，那么应该尽量将数据进行适当的分组。

加法和减法的另一个问题是可能最终得到错误的精度。考虑运算1.23e0–1.22e0。它将产生结果0.01e0。虽然这在算术上等效于1.00e–2，但是后面这种格式表示后面两位数字精确为0。遗憾的是，此时我们只得到了一个有效的数字。事实上，某些浮点单元(floating-point unit，FPU)软件包可能会向低位插入随机数字(或者位)。这就又产生了一条关于有限精度运算的重要规则：

两个同号的数相减或者两个异号的数相加时，结果的精度可能小于浮点格式中所能够

得到的精度。

乘法和除法不存在与加法和减法同样的问题，因为在操作以前不需要对它们的指数进行调整；需要做的是将指数相加，将尾数相乘(或者将指数相减，将尾数相除)。乘法和除法本身并不会产生精度特别低的结果。但是，它们往往会使得数值中已存在的错误加倍。例如，如果应该将1.24e0乘以2，而您将1.23e0乘以2，那么结果就更加不正确了。这就引入了有限精度运算的又一条重要规则：

在执行一系列涉及加减乘除的运算中，应该先执行乘法和除法运算。

通常，通过应用标准的代数变换就可以安排运算的顺序，使得乘法和除法首先进行。例如，假设要计算x*(y+z)。正常顺序是将y与z相加，再用它们的和乘以x。但是如果对x*(y+z)进行变换得到x*y+x*z，并通过先执行乘法来计算结果，那么就会得到更高的精度6。

并不是乘法和除法本身就不存在问题。将两个非常大或者非常小的数相乘时，很可能会发生上溢或者下溢。当用一个很大的数除以一个很小的数或者用一个很小的数除以一个很大的数时也会出现同样的情况。这就引入了在进行乘法和除法操作时应该尽量遵循的第4条规则：

当对一组数值进行乘法操作和除法操作时，应该尽量安排大数与小数相乘；同样，尽

量安排两个具有相同数量级的数值相除。

对浮点数进行比较操作是非常危险的。考虑到任何一个计算都会出现错误(包括将一个输入字符串转换成浮点数值)，所以千万不要通过两个浮点数进行比较来判断它们是否相等。在二进制浮点格式中，产生相同(算术)结果的不同运算可能会在最低有效位上发生差异。例如，做加法1.31e0+1.69e0应该得到3.00e0。同样，做加法1.50e0+1.50e0应该得到3.00e0。但是，如果对(1.31e0+1.69e0)和(1.50e0+1.50e0)进行比较就会发现这两个和是互不相等的。当且仅当两个操作数中所有的位(或者数字)都完全相同时，相等性检查才会成功。因为在两个应该产生相同结果的不同的浮点运算执行完以后，这一点并不一定会成立，直接测试是否相等可能不会产生正确的结果。

检查两个浮点数是否相等的标准方法是确定在比较中允许存在的误差范围(允差范围)，并检查其中一个数值是否在另一个数值的误差范围内。通常的方法就是进行如下所示的检查：

if Value1 >=( Value2-error )  and Value1 <= (Value2+error)  then ...

另一种处理这种比较的常用方法就是使用如下格式的语句：

if abs(Value1-Value2) < =error then ...

在选择error值时一定要仔细。它应该是比运算中可能出现的最大误差值稍小的数。具体数值取决于所使用的浮点格式，稍后将作详细的阐述。我们要提出的最后一条规则是：

在对两个浮点数进行比较的时候，总是对一个数值进行比较，看它是否处于第二个数值加上或者减去一个小的误差值所得到的范围之内。

在使用浮点数的过程中，还可能发生很多其他的小问题。本书只能指出其中一些主要问题，让您知道不能像对待实数运算一样对待浮点运算这样一个事实—— 如果不仔细，那么有限精度运算中出现的不准确度将会带来很大的麻烦。关于数字分析或者科学计算的好文章都有助于补充超出本书范围的一些细节知识。如果要使用浮点运算，就应该花时间对运算过程中的有限精度运算所产生的影响进行研究。

HLA的if语句不支持涉及浮点操作数的布尔表达式。因此，在程序中不能使用类似于if( x < 3.141) then . . .这样的语句。第6章将介绍如何进行浮点数比较。

2.12.1  IEEE浮点格式

当Intel打算在它的新8086微处理器中引入浮点单元(FPU)时，它非常明智地认识到，在为浮点格式选择一种尽可能最好的二进制表示时，设计芯片的电子工程师和固体物理学家可能并不是最佳人选。所以Intel公司聘请了一位最好的数字分析专家来为8087 FPU设计浮点格式。然后这个专家又聘请了该领域中的另外两位专家，他们三人(Kahn、Coonan和Stone)共同设计了Intel的浮点格式。他们干得非常漂亮，设计了KCS浮点标准，使得IEEE组织将该格式作为IEEE浮点格式7。

为了满足各种性能和精度需求，Intel实际上引入了3种浮点格式：单精度、双精度以及扩展精度。单精度和双精度格式分别对应于C语言中的float和double类型或FORTRAN中的real和double precision类型。Intel想在较长的运算链中使用扩展精度类型。扩展精度类型包含16个额外的位，在存储结果的时候，运算可以在四舍五入到双精度之前将这些额外的位用作保护位。

单精度格式用到一个由反码表示的24位尾数和一个8位的excess-127格式的指数。尾数通常为1.0~2.0之间的数值。尾数的最高位总是被假设为1，表示正好位于二进制小数点(binary point)左边的数值8。余下的23位尾数位于二进制小数点的右侧。因此，尾数表示为：

1.mmmmmmm mmmmmmmm mmmmmmmm

mmmm . . .字符代表尾数中的其他23位。记住，我们在这里处理的是二进制数。因此二进制小数点右边的每一个位置表示一个数值(0或1)乘以2的负逐次幂。隐含的那一位总是乘以20，即1。这就是为什么尾数总是大于或者等于1的原因。即使其他的尾数位都为0，隐含的那一位也总是数值19。当然，即使在二进制小数点后面存在无限个1，它们相加仍然不能得到2。这就是为什么尾数能够表示1~2范围内的值的原因。

虽然在1~2之间存在无限个数值，但我们只能对其中的800万个数进行表示。因为我们使用的是23位的尾数(第24位总是1)。这就是浮点运算不精确的原因——在涉及单精度浮点数值的运算中，精度被限制为23位。

尾数采用的是反码格式而不是补码格式。这就意味着尾数的24位数值只是一个无符号二进制数，并且符号位决定了该数值是正还是负。反码数具有不寻常的属性，它对0有两种表示(符号位被设置或者清除)。一般情况下，这只对那些设计浮点软件或者硬件系统的人来说很重要。我们假设数值0总是将符号位清零。

当要表示1.0~2.0这个范围以外的数值时，浮点格式的指数部分才能发挥作用。浮点格式将2自乘到指数指定的次幂，再用该数值与尾数相乘。指数为8位，以excess-127格式存储。此时，指数20用数值127($7F)表示。因此，要将一个指数转换为excess-127格式，只需要简单地将127和指数值相加。使用excess-127格式简化了浮点数的比较。单精度浮点格式采用图2-22所示的形式。

图2-22  单精度(32位)浮点格式

采用24位尾数所得到的精度近似于

精度的意思是前6个数字可以在范围0~9之间，但是第7个数字只能在范围0~x之间，其中x<9且一般都接近5)。采用8位的excess-127格式的指数时，单精度浮点数的动态范围大约为

虽然单精度浮点数适合于很多情况，但是动态范围却受到一些限制，且不适合于金融、科学以及很多其他的应用。而且，在进行长链运算的时候，单精度浮点格式的有限精度可能会引入很严重的错误。

双精度浮点格式能够克服单精度浮点数的缺点。双精度浮点格式占用了两倍的空间，具有一个11位的excess-1023格式的指数和一个53位的尾数(带一个隐含为1的高位位)再加上一个符号位。这提供了一个大约为10±308的动态范围和14

精度的数字，它满足大多数的应用。双精度浮点数值采用图2-23所示的格式。

图2-23  64位双精度浮点格式

在涉及到双精度浮点数的长链运算时，为了确保精度，Intel设计了扩展精度格式。扩展精度格式需要占用80位。附加的16位中有12位被附加到尾数上，另外4位被添加到指数部分的末尾。与单精度和双精度数值不同，扩展精度格式的尾数不包含隐含的且总是为1的高位位。因此扩展精度格式提供了一个64位的尾数、一个15位的excess-16383格式的指数以及一个1位的符号位。扩展精度浮点数值的格式如图2-24所示。

图2-24  80位扩展精度浮点格式

在FPU上进行的所有运算都采用扩展精度格式。只要加载一个单精度或者双精度的数值，FPU就会自动将它转换成一个扩展精度的数值。同样，只要向存储器中存入一个单精度或者双精度的数值，FPU就会在存入以前自动将该数值四舍五入到适当的长度。Intel公司总是利用扩展精度格式进行操作，因而保证了引入大量的保护位来确保运算的精度。

为了在运算中维持最高的精度，大多数运算都使用规格化的数值。规格化浮点数值是那些尾数最高位为1的数。几乎所有的非规格化数都能够被规格化；将尾数的各位向左移，减小指数直到尾数的高位位出现1为止。记住，指数是二进制的。每增加一次指数，就是将浮点数值乘以2。同样，每减小一次指数就是将浮点数值除以2。同理，将尾数左移一个位置就要将浮点数乘以2；将尾数向右移就要将浮点数除以2。因此，将尾数左移一个位置并将指数减1根本不会改变浮点数的数值。

使浮点数规格化是有好处的，因为它能保证运算结果最大的精度。如果尾数的高位位都是0，那么尾数为运算保持的精确位就要少得多了。因此，如果一个浮点运算只包含规格化的数值，那么它就会更加精确。
有两种重要的情况，浮点数不能规格化。0是特殊情况之一。0不能被规格化，是因为它的浮点表示中尾数不含值为1的位。但是，这并不是问题，因为我们可以只用一位来表示数值0。

第二种情况就是当尾数的高位位为0但指数也为0(我们不能减小它来对尾数进行规格化)的时候。对于尾数的高位位和指数都为0的那些很小的数，IEEE标准并没有不接受它们，而是用一些特殊的非规格化数来表示10。虽然与发生下溢的情况相比较，使用非规格化数值可以使IEEE浮点运算产生更准确的结果，但是它提供的精确位更少。

2.12.2  HLA为浮点数值提供的支持

HLA提供了几种数据类型和库例程来支持汇编语言程序中浮点数据的使用，包括用于声明浮点变量的内置类型以及提供浮点输入、输出以及转换的例程。

在对HLA提供的浮点支持进行讨论时，最好从浮点字面常量开始。HLA浮点常量采用如下所示的语法：

· 一个可选的+号或–号，代表尾数的符号(如果这个符号没有出现，HLA就假定尾数为正)

· 后面跟着一位或者多位十进制数

· 后面跟着一个小数点和一位或者多位十进制数(本项是可选项)
· 后面跟着一个e或者E，再可选地跟着一个符号位(+或者–)和一位或者多位十进制数(本项是可选项)
注意，为了将该数值与整数或者无符号字面常量进行区分，必须出现小数点或者e/E。下面是一些合法的浮点字面常量：

1.234  3.75e2  -1.0  1.1e-1  1e+4  0.1  -123.456e+789  +25e0

注意浮点字面常量不能从小数点开始；它必须以十进制数字开头，所以在程序中必须用“0.1”来表示“.1”。

HLA还允许在浮点字面常量中两个连续的十进制数字之间插入一个下划线符号“_”。可以用下划线符号代替逗号(或者其他的语言指定的分隔符)以使得较长的浮点数更容易阅读。下面是一些例子：

1_234_837.25  1_000.00  789_934.99  9_999.99

可以用数据类型real32、real64、或者real80来声明浮点变量。像整数和无符号数值一样，这些数据类型声明末尾的那个数指定了每种类型的二进制表示所使用的位数。因此，real32用于声明单精度实数，real64用于声明双精度浮点数，real80用于声明扩展精度浮点数。除了使用这些类型来声明浮点数而不是整数这一事实以外，它们的用法基本与int8、int16、int32等相同。下面的例子展示了这些声明以及它们的语法：
static

fltVar1:












real32;

fltVar1a:











real32 := 2.7;

pi:
















real32 :=







3.14159;

DblVar:












real64;

DblVar2:












real64 := 1.23456789e+10;

XPVar:













real80;

XPVar2:












real80 := -1.0e-104;
要以ASCII格式输出浮点变量，可以使用例程stdout.putr32、stdout.putr64或者stdout.putr80。这些过程以小数点表示法显示数值—— 也就是说，一个数字串、一个可选的小数点再加上一个数字串结尾。除了它们的名称以外，这三个例程使用完全相同的调用序列。下面显示了对这些例程的调用方法以及相应参数：

stdout.putr80( r:real80; width:uns32; decpts:uns32 );

stdout.putr64( r:real64; width:uns32; decpts:uns32 );

stdout.putr32( r:real32; width:uns32; decpts:uns32 );
这些过程的第一个参数为要打印的浮点数值。该参数的长度必须与过程的名称相匹配(例如，当调用stdout.putr80时，参数r必须是一个80位的扩展精度浮点变量)。第二个参数为输出文本指定了域宽；这是过程显示数据时所需要的打印位置数。注意，该宽度必须包含数据的符号位和小数点的打印位置。第三个参数指定了小数点后面的数所需要的打印位置。例如：

stdout.putr32( pi, 10, 4 );

显示数值：

_ _ _ _ 3.1416

在本例中下划线表示前导空格。

当然，如果该数非常大或者非常小，那么在浮点数输出时就会想到采用科学记数法而不是小数点表示法。HLA标准库例程stdout.pute32、stdout.pute64以及stdout.pute80为此提供了方便。这些例程采用的过程原型如下所示：

stdout.pute80( r:real80; width:uns32 );

stdout.pute64( r:real64; width:uns32 );

stdout.pute32( r:real32; width:uns32 );
与小数点输出例程不同的是，这些科学记数法输出例程不需要第三个参数来指定小数点后面要显示的数字的位数。参数width间接指定了该数值，因为所有的尾数位当中总是只有一位出现在小数点左边。这些例程以小数点表示法输出它们的数值，与下面所示的类似：

1.23456789e+10     -1.0e-104     1e+2

还可以使用HLA标准库中的stdout.put例程来输出浮点数值。如果在stdout.put的参数列表中指定一个浮点变量的名称，stdout.put程序就会采用科学记数法输出该数值。实际的域宽取决于浮点变量的长度(在这种情况下，stdout.put例程会尝试尽可能多地输出有效数字)。例如：

stdout.put( "XPVar2 = ", XPVar2 );

如果使用一个冒号并在后面给出一个有符号整数值来指定域宽，那么stdout.put例程就会调用适当的stdout.puteXX例程显示该数值。也就是说，该数仍然会以科学记数的形式出现，但是可以控制输出数值的域宽。与整数和无符号数的域宽一样，域宽为正数表明指定域中的数为右对齐，域宽为负数说明指定域中的数为左对齐。这里给出了一个使用10个打印位置打印变量XPVar2的例子：

stdout.put( "XPVar2 = ", XPVar2:10 );

如果想用stdout.put以小数形式打印浮点数，就需要使用下面所示的语法格式：

Variable_Name : Width : DecPts
注意，DecPts域必须是非负的整数值。

当stdout.put中包含这种格式的参数时，它就会调用相应的stdout.putrXX例程显示指定的浮点数值。作为例子，考虑如下所示的调用：

stdout.put( "Pi = ", pi:5:3 );

其对应的输出为：

3.142

HLA标准库提供了其他几种有用的例程，可以在输出浮点数值的时候使用这些例程。关于这些例程的更多信息可以参阅HLA标准库参考手册。

HLA标准库提供了几种例程，它们可以采用不同格式显示浮点数据。与之相比，HLA标准库只提供了两种例程支持浮点输入：stdin.getf()和stdin.get()。stdin.getf()例程需要用到80x86的FPU栈，它是一个硬部件，本章不准备对它进行讨论。因此我们把对stdin.getf( )的讨论推迟到第6章进行。因为stdin.get()例程提供了stdin.getf()例程具备的所有功能，所以这样推迟以后不会出现什么问题。

您已经看到了stdin.get()例程的语法；它的参数列表只包含一个变量名列表。stdin.get()函数为用户读入参数列表中出现的每个变量的数值。如果浮点变量的名称已经指定了，那么stdin.get()例程就会自动从用户处读入浮点数值，并将结果存入指定变量中。下面的例子展示了该例程的应用：

stdout.put( "Input a double precision floating-point value:" );
stdin.get( DblVar );

警告：
本节已经讨论了如何声明浮点变量，以及如何对它们进行输入和输出，但没有对算术运算进行讨论。浮点算术运算与整数的算术运算是不同的；不能使用80x86的add和sub指令来对浮点数值进行操作。浮点算术运算将在第6章讲述。

2.13  BCD数据表示

虽然整数和浮点格式已经可以满足一般程序的数据表示需求，但是在一些特殊的情况下，其他的数字表示方式更加方便一些。在本节中，我们将对二进制编码的十进制(BCD)格式进行讨论，因为80x86 CPU为这种数据表示方式提供了少量的硬件支持。

BCD数值是由一些半字节组成的序列，每个半字节表示0~9中的某个数值。当然，也可以使用半字节来表示0~15中的数值；但是在每个半字节所能表示的16个不同数值中，BCD格式只使用了其中的10个。

BCD数值中的每个半字节表示一个十进制数字。因此，我们用一个字节(也就是两个数位)可以表示包含两个十进制数字的数值或者0~99内的数值(见图2-25)。用一个字，我们可以表示具有四个十进制数字的数值或者说0~9999内的数值。同样，使用一个双字，我们可以表示具有八个十进制数字的数值(因为在一个双字数值当中存在八个半字节)。

图2-25  存储器中BCD数据表示

正如您所看到的，BCD存储并没有有效利用存储器。例如，一个8位的BCD变量可以表示0~99内的数值，然而当保存二进制数值的时候，同样的8位可以表示0~255内的数值。同样，16位的二进制数值可以表示0~65535内的数值，而一个16位的BCD数值只能表示这些数值中的1/6(0~9999)。存储效率低还不是唯一的问题。BCD的计算往往比二进制的计算更慢。

在此，您可能想知道为什么会有人采用BCD格式。BCD格式确实存在两个优点：BCD数值的内部数字表示和它们的字符串表示之间进行转换是非常容易的；而且，在用硬件(例如，使用微型旋轮或者刻度盘)完成多位十进制数值的编码时，使用BCD码比使用二进制要容易得多。正是由于这两个原因，您很可能会在嵌入式系统(例如，微波炉、闹钟和核反应堆)中看到人们使用BCD码，但是却很少看到它们应用于一般的计算机软件中。

几十年以前，人们错误地认为涉及BCD码(或者只是“十进制”)的算术运算比二进制运算更加精确一些。因此，他们通常都采用基于十进制的算术运算来执行那些重要的计算，比如美元和美分(或者其他的金融单位)的计算。虽然某些采用BCD码的计算能够产生更加精确的结果，但是这种说法在一般情况下是不正确的。事实上，对于大多数计算(甚至那些涉及定点十进制的算术运算)来说，二进制表示更加精确一些。因此，现在大多数的计算机程序都采用二进制格式来表示所有的数值。例如，Intel的80x86浮点单元(FPU)对两条用于存储和取出BCD数值的指令提供了支持。然而在内部，FPU将这些BCD数值转换为二进制表示，并采用二进制来执行所有的计算。它只将BCD码用作外部数据格式(也就是说，在FPU的外面)。这样一般就能产生更加精确的结果，与专门拥有一个能支持十进制计算的协处理器相比，它占用的硅片面积更小。

2.14  字符

在个人计算机上最重要的数据类型可能就是字符类型。“字符”指的是人或者机器可读的符号，一般都是非数字实体。一般来说，“字符”指的是在键盘上能够正常输入(包括那些需要多个按键才能得到的)或者在显示器上能够显示的字符。很多初学者通常会对“字符”与“字母字符”产生混淆。这些术语是不一样的。标点符号、数字符号、空格符、制表符、回车符、其他的控制字符以及特殊字符也都属于字符。当本书中用到术语“字符”时，指的是这些字符中的任意一种，而不只是字母字符。当提到术语“字母字符”时，本书会使用像“字母字符”、“大写字符”或者“小写字符”之类的术语。

当初学者第一次遇到字符数据类型时，另一个普遍存在的问题是数字字符与数字之间的区别。字符1与数值1是不同的。计算机使用两种不同的内部表示来表示数字字符(0、1、…、9)和数值0~9。一定要小心，不要将两者混淆。

大多数计算机系统都使用单字节或者双字节序列将各种不同的字符编码为二进制形式。Windows、Mac OS X、FreeBSD和Linux也属于这一类，使用ASCII码或者Unicode来对字符进行编码。本部分将对HLA提供的ASCII字符集和字符声明方法进行讨论。

2.14.1  ASCII字符编码

ASCII(American Standard Code for Information Interchange，美国信息交换标准码)字符集将128个文本字符映射为无符号整数值0~127($0~$7F)。当然，所有的对象在计算机内部都采用二进制数来表示，所以计算机用二进制数值来表示像字符这样的非数字实体就不值得惊讶了。虽然字符到数字的映射是任意的，并且不太重要，但是使用标准化的代码来表示这种映射是很重要的，因为需要与其他程序和外设进行通信，且需要与这些程序和设备使用相同的“语言”。这就该ASCII代码发挥作用了；它是一种已经达成一致的标准化代码。因此，如果用ASCII代码65来表示字符A，那么就会知道无论何时将这一数据传递给外设(如打印机)，它们都会将该数值解释为字符A。

不应该认为计算机系统中只使用ASCII一种字符集。IBM在它的很多大型机系统中都使用EBCDIC字符集。另一种通用字符集是Unicode字符集。Unicode是对ASCII字符集的扩展，它采用16位而不是7位来表示字符。这就允许在字符集中使用65536个字符，将世界上不同语言的大部分符号包含到一种统一的字符集中。

因为ASCII字符集只提供了128个不同的字符，而一个字节可以表示256个不同的数值，这样就产生了一个很有趣的问题：“我们应该如何处理存储到一个字节中的数值128~255？”一种方法是忽略那些额外的数值。这是本书采用的最主要的方法。另一种可能性是扩展ASCII字符集，向字符集中加入128个字符。当然，这往往不能得到一种标准化的字符集，除非您能让所有人都赞成这种扩展。这是一项很困难的任务。

当IBM首次生产出IBM-PC时，定义了这些额外的128个字符代码来保存各种非英文字母字符、由线画成的字符、数学符号以及其他几种特殊字符。因为IBM公司的PC是我们今天所说的一般PC的基础，所以它的字符集已经成为所有与IBM-PC兼容的机器的一种伪标准。甚至在现代的一些与IBM-PC不兼容且不能运行早期PC软件的机器上，IBM的扩展字符集仍然存在。但是，要注意该PC字符集(ASCII字符集的扩展)并不是通用的。当使用自带的字体时，大多数打印机都不会打印扩展字符，并且很多程序(尤其是在非英语的国家)都不在8位数值中使用扩展字符来表示额外的128个字符代码。正是因为这些原因，本书一般都坚持使用标准的128字符的ASCII字符集。

尽管事实上它是一个标准，但如果只是简单地使用标准的ASCII字符对数据进行编码，并不能保证各个系统间的兼容性。虽然大多数情况下，一台机器上的A很可能就是另一台机器上的A，但是关于控制字符的使用，不同机器之间的标准化非常少。事实上，32个控制代码加上DELETE，只有4个控制代码得到了普遍的支持——空格(BS)、制表符、回车符(CR)以及换行符(LF)。更糟糕的是，不同机器通常采用不同的方式来使用这些控制代码。行末是一个特别麻烦的例子。Windows、MS-DOS、CP/M以及其他系统采用两字符的序列CR/LF来对行末进行标记。Apple的Macintosh和很多其他系统只用一个CR字符来标记行末。Linux、Mac OS X、FreeBSD以及其他UNIX系统使用单个LF字符来对行末进行标记。无须多说，要在这些系统之间交换文件会遇到很多麻烦。即使这些系统上的所有文件采用的都是标准的ASCII字符，在它们之间交换文件时仍然需要对数据进行转换。幸运的是，这样的转换都十分简单。

尽管ASCII数据还存在一些主要的缺点，它仍然是计算机系统以及程序之间进行数据交换的标准。大多数程序都能够识别ASCII数据；同样，大多数程序都能够生成ASCII数据。因为要使用汇编语言来处理ASCII字符，所以需对字符集的编排进行研究，记住几个关键的ASCII代码(例如，0、A、a等)是比较明智的。

ASCII字符集被分成4组，每组由32个字符组成。前32个字符，即ASCII码0~$1F(31)构成了非打印字符(即控制字符)的特殊字符集。我们之所以把它们称为控制字符，是因为它们执行的是各种打印/显示的控制操作，而不是显示符号。这一类例子包括回车(将光标放到字符所在行的左边)11、换行(将光标移到输出设备的下一行)以及退格(将光标向左移动一个位置)。遗憾的是，不同的控制字符在不同的输出设备上执行不同的操作。在这些输出设备之间存在的统一标准很少。要想了解某个控制字符是如何影响某个特定设备的，需要参阅它的用户手册。

第2组的32个ASCII字符代码由各种标点符号、特殊字符以及数字组成。在该组中最值得注意的字符包括空格符(ASCII码为$20)和数字(ASCII码为$30~$39)。

第3组的32个字符包含所有的大写字母字符。字符A~Z的ASCII码位于$41~$5A(65~90)之间。由于只有26个不同的字母字符，所以余下6个代码就用于表示特殊符号。

第4组，即最后一组32个ASCII码表示小写字母字符、5个特殊符号以及一个控制符(DELETE)。注意，小写字母字符使用$61~$7A之间的ASCII码。如果将大写和小写字母字符转换成二进制，您就会注意到大写字母字符的ASCII码与它们所对应的小写字母恰好有一位不同。例如，考虑图2-26中给出的E和e的字符代码。

图2-26  E与e的ASCII码

这两个代码唯一不同的位置就是第5位。大写字母字符的第5位总是0；而小写字母字符的第5位总是1。这样就可以利用这一事实快速实现大小写之间的转换。如果有一个大写字母，可以通过把第5位设置为1将它强制转换为小写字符。如果有一个小写字母，并且想将它强制转换成大写形式，那么可以通过将第5位设置为0实现。因此，只需要通过转换第5位就可以实现字母字符的大小写转换了。

事实上，如表2-8所示，第5位和第6位就决定了字符处于4组ASCII字符集中的哪一组。

表2-8  ASCII组

	第6位
	第5位
	组

	0
	0
	控制字符

	0
	1
	数字字符和标点符号

	1
	0
	大写字符和特殊字符

	1
	1
	小写字符和特殊字符


例如，可以通过把第5位和第6位都设置为0，将任一大写或者小写(或者相应的特殊)字符转换成等效的控制字符。

考虑表2-9中给出的数字字符的ASCII码。

表2-9  数字字符的ASCII码

	字    符
	十  进  制
	十 六 进 制

	0
	48
	$30

	1
	49
	$31

	2
	50
	$32

	3
	51
	$33

	4
	52
	$34

	5
	53
	$35

	6
	54
	$36

	7
	55
	$37

	8
	56
	$38

	9
	57
	$39


这些ASCII码的十进制表示并没有很大的启发性。但是，这些ASCII码的十六进制表示却反映了一些非常重要的问题：ASCII码的低位半字节与它所表示的数字的二进制表示相同。只要把某个数字字符的高位半字节去掉(也就是说，设置为0)，就可以将该字符代码转换为与其对应的二进制表示。相反，只要把高位半字节设置成3就可以将0~9内的某个二进制数值转换为它的ASCII码。注意，这里可以使用逻辑与操作将高位位强制设为0；同样，可以使用逻辑或操作将高位位强制设为%0011(3)。

注意，在数字字符串中，只去掉其中每一个数字的高位半字节不能将该数字串转换为它们等效的二进制表示。如果以这种方式来转换123($31 $32 $33)就可以得到三个字节：$010203；而123对应的正确的数值应该为$7B。如果要将数字串转换为整数就比这更加复杂了；上面所使用的转换方法只适用于单个的数字。

2.14.2  HLA对ASCII字符提供的支持

虽然很容易将字符数值存储到字节变量中，并且在程序中使用字符字面量时，也很容易使用其相应的ASCII码，但是这种痛苦是没必要的：HLA为汇编语言程序中的字符变量和字面量提供了很好的支持。

HLA中的字符字面常量可采用两种形式来表示：由单引号括起来的单个字符，或者在一个“#”符号后面跟上用于指定该字符的ASCII码的数字常量，该数字常量处于范围0~127之间。这里有一些例子：

'A'    #65    #$41    #%0100_0001

注意，这些例子都表示同一个字符“A”，因为“A”的ASCII码为65。

除了一种例外的情况以外，在字面字符常量中，单引号之间只能出现一个字符。这种例外就是单引号字符本身。如果想创建一个单引号字面常量，那么可以通过在一行中放置四个单引号来实现(也就是说，将单引号用单引号括起来)：

''''
“#”操作符必须放在合法的HLA数字常量(正如上面例子当中所指出的，可以是十进制、十六进制或者二进制)的前面。特别要注意的是，“#”符号不是通用的字符转换函数；它不能放在寄存器或者变量名的前面，只能放在常量前面。

按照常规，应该总是使用字符字面常量的单引号形式来处理图形字符(也就是说，那些可以打印或者可以显示的字符)。采用“#”符号的格式来处理控制字符(它们在打印时不可见或者会实现一些有趣的功能)或者源代码中出现的不能正确显示和打印的扩展ASCII字符。

注意程序中的字符字面常量与字符串字面常量之间的区别。字符串是由0个或者多个用双引号括起来的字符组成的序列；而字符是用单引号括起来的。

认识到‘A’

“A”尤其重要。字符常量A与包含单个字符A的字符串的内部表示完全不同。如果在HLA希望出现字符常量的地方使用包含一个单字符的字符串，HLA就会报告出错。本书将在第4章专门介绍关于字符串和字符串常量的内容。

要在HLA程序中声明一个字符变量，可以使用char数据类型。例如，下面所示的声明展示了如何声明一个名为UserInput的变量：

static
UserInput:   char;

该声明保留一个字节的存储空间，可以保存任一字符数值(包括8位的扩展ASCII字符)。也可以对字符变量进行初始化，如下所示：

static

TheCharA:







char := 'A';
ExtendedChar:




char := #128;

因为字符变量是一个8位的对象，所以可以使用一个8位寄存器来对它们进行操作。可以将字符变量移到8位寄存器中，也可以将8位寄存器的数值存储到字符变量中。

HLA标准库提供了一些例程，使用这些例程可以完成字符的输入输出和处理；这些例程包括stdout.putc、stdout.putcSize、stdout.put、stdin.getc以及stdin.get。

stdout.putc例程使用了下面所示的调用序列：

stdout.putc( charvar );

该过程将一个作为字符传递给它的单字符输出到标准输出设备。该参数可以是任意的char常量、变量、字节变量或者寄存器12。

stdout.putcSize例程对显示字符变量时的输出宽度提供了控制。该过程的调用序列为
stdout.putcSize( charvar, widthInt32, fillchar );

该例程使用至少等于widthInt32的输出位数输出指定的字符(参数c) 13。如果widthInt32的绝对值大于1，那么stdout.putcSize将输出fillchar作为填充字符。如果widthInt32为正数，那么stdout.putcSize将在输出域中右对齐输出字符；如果width为负数，stdout.putcSize将在输出域中左对齐输出字符。由于输出的字符通常是左对齐的，所以在该程序调用中，widthInt32的值通常为负。空格符是常用的填充字符。

也可以使用stdout.put例程输出字符值。如果某个字符变量出现在其参数列表中，则stdout.put自动将其作为字符值输出，例如：

stdout.put( "Character c = '", c, "'", nl );

采用stdin.getc和stdin.get例程可以从标准输入设备中读入字符。stdin.getc程序没有任何参数。它从标准输入缓冲区中读入单个字符并将该字符返回AL寄存器。然后就可以将该字符存往别处或者在AL寄存器中操作该字符。程序清单2-10中的程序从用户端读入一个单字符，如果为小写，就将其转换为大写形式，然后输出该字符。

程序清单2-10  字符输入示例

program charInputDemo;

#include( "stdlib.hhf" )

begin charInputDemo;

stdout.put( "Enter a character: " );

stdin.getc();

if( al >= 'a' ) then

if( al <= 'z' ) then

and( $5f, al );

endif;

endif;

stdout.put

(

"The character you entered, possibly ", nl,

"converted to uppercase, was '"

);

stdout.putc( al );

stdout.put( "'", nl );

end charInputDemo;
还可以使用stdin.get例程从用户端读入字符变量。如果某个stdin.get参数是字符变量，该例程将从用户端读入一个字符并将其值存入指定的变量。程序清单2-11使用stdin.get例程重写程序清单2-10。

程序清单2-11  stdin.get字符输入示例

program charInputDemo2;

#include( "stdlib.hhf" )

static

c:char;

begin charInputDemo2;

stdout.put( "Enter a character: " );

stdin.get(c);

if( c >= 'a' ) then

if( c <= 'z' ) then

and( $5f, c );

endif;

endif;

stdout.put

(

"The character you entered, possibly ", nl,

"converted to uppercase, was '",

c,

"'", nl

);

end charInputDemo2;
回想第1章，HLA标准库将输入内容存入缓冲区。不管何时利用stdin.getc或stdin.get从标准输入中获得一个字符，库例程都会读入该缓冲区中下一个可用的字符；如果缓冲区为空，程序读入用户(输入的)新的一行文本并返回该行的首字母。若要保证在读入某个字符变量时，程序从用户端读入新的一行文本，在试图读入数据之前，应当调用stdin.flushInput例程。这将刷新当前的输入缓冲区并强迫在下一次输入的时候读入新的一行文本(可能是一次stdin.getc或者stdin.get调用)。

行末存在问题。不同的操作系统以不同的方式处理输入的行末和输出的行末。从控制台设备来看，按下ENTER键就意味着一行的结束；但是，当从文件读入数据时，得到的行尾序列是换行或回车/换行对(在Windows下)，或者仅仅是换行(在Linux/Mac OS X/FreeBSD下)。为了解决该问题，HLA的标准库提供了一个“行尾”函数。如果当前所有的输入字符都用完了，该程序将true(1)返回AL寄存器中，反之返回false(0)。程序清单2-12的示例程序演示了stdin.eoln函数。

程序清单2-12  使用stdin.eoln测试某行的末尾

program eolnDemo;

#include( "stdlib.hhf" )

begin eolnDemo;

stdout.put( "Enter a short line of text: " );

stdin.flushInput();

repeat

stdin.getc();

stdout.putc( al );

stdout.put( "=$", al, nl );

until( stdin.eoln() );

end eolnDemo;
HLA语言和HLA标准库对字符对象提供了许多其他的过程以及附加的支持。本书的第4章和第11章以及HLA参考文档介绍了如何使用这些特性。

2.15  Unicode字符集

尽管ACSII字符集毫无疑问是计算机上最流行的字符表示，但它不是唯一的格式。例如，在许多大型机和小型机系列中，IBM使用EBCDIC码。因为EBCDIC主要出现在IBM的大型计算机上，而在个人计算机系统上很少碰到，本书不考虑该字符集。另一种字符表示在小型计算机系统上变得流行起来，这就是Unicode字符集。Unicode克服了ASCII码的两个最大限制：对字符空间的限制(最多只能有128/256个8位字符)和缺少国际字符。

Unicode使用16位的字来表示单个字符。因此，它支持65536个不同的字符码。与使用8位字节可表示的256个字符码相比，这显然是一个极大的进步。Unicode与ASCII是向上兼容的。具体来说，如果一个Unicode字符的高9位含0，则其低7位表示的字符和用相同字符编码表示的ASCII字符相同。如果高9位含有非零值，该字符表示其他的含义。如果想知道为什么需要这么多不同的字符码，可以注意一下，一些亚洲字符集含有4096个字符(至少它们的Unicode子集是这样)。

除了简单提到Unicode的地方，本书将一直使用ASCII字符集。不过最终，本书还是必须撇下有关ASCII的讨论而转向讨论Unicode，因为许多新的操作系统在内部使用Unicode (必要时转换为ASCII)。遗憾的是，许多字符串算法对于Unicode没有ASCII那么方便(特别是字符集函数)，因此本书尽可能坚持使用ASCII码。

2.16  更多信息

本书完整的电子版(可从http://webster.cs.ucr.edu/或http://artofasm.com/站点上下载)含有一些额外的数据表示信息，可能对您有用。若要了解数据表示的详细内容，可以参考笔者编写的Write Great Code, Volume 1(No Starch Press,2004)一书，或者其他有关数据结构和算法方面的书。

1 二进制编码的十进制数是一种用来表示十进制数的方法，它将每一位十进制数字都用4位来表示。


2 很多书都把它叫做“二元(binary)”运算。“二元(dyadic)”的意思与之相同，避免了与二进制数字系统发生混淆。


3 这并不能算作是一种限制，因为符号扩展总是在寄存器中的算术操作完成之前进行。


4 最终您会明白，零扩展为DX:AX或者EDX:EAX与使用cwd和cdq指令一样必要。


5 算术左移并不需要。假如没有上溢发生，标准的左移操作对有符号数和无符号数都适用。


6 当然，缺点是现在必须执行两个乘法而不是一个乘法，所以结果会更慢。


7 处理某些退化运算时所采用的方法有一些小的变化，但是位表示基本上没变。


8 二进制小数点与十进制小数点是相同的，只是它出现在二进制数当中而不是十进制数中。


9 事实上，这并不一定对。IEEE浮点格式支持高位位不为0的非规格化(denormalized)数。但是在我们的讨论当中，将忽略非规格化的数值。


10 另一种办法是将数值下溢为0。


11 在历史上，回车指的是打印机上面所用的纸架(paper carriage)。回车操作就是将纸架整个都移动到右边，使得要打印的下一个字符出现在纸的左边。


12 如果参数被指定为字节变量或者字节大小的寄存器，那么stdout.putc例程将输出变量或者寄存器中给出的ASCII码所对应的字符。


13 只有在指定宽度为0的时候，stdout.putcSize才使用更多的输出位数；这时例程使用的输出位数刚好为1位。

